
Abstract

Differential geometry is deceptively simple. It is surprisingly easy
to get the right answer with unclear and informal symbol ma-
nipulation. To address this problem we use computer programs
to communicate a precise understanding of the computations in
differential geometry. Expressing the methods of differential ge-
ometry in a computer language forces them to be unambiguous
and computationally effective. The task of formulating a method
as a computer-executable program and debugging that program
is a powerful exercise in the learning process. Also, once formal-
ized procedurally, a mathematical idea becomes a tool that can
be used directly to compute results.

A manifold is a generalization of our idea of a smooth sur-
face embedded in Euclidean space. The critical feature of an
n-dimensional manifold is that locally (near any point) it looks
like n-dimensional Euclidean space: around every point there is a
simply-connected open set, the coordinate patch, and a bijective
continuous function, the coordinate function or chart mapping ev-
ery point in that open set to a tuple of n real numbers, the coordi-
nates. In general, several coordinate systems are needed to label
all points on a manifold. It is required that if a region is in more
than one coordinate patch then the coordinates are consistent in
that the function mapping one set of coordinates to another is
continuous (and perhaps differentiable to some degree). A consis-
tent system of coordinate patches and coordinate functions that
covers the entire manifold is called an atlas.1

An example of a two-dimensional manifold is the surface of a
sphere or of a coffee cup. The space of all configurations of a planar
double pendulum is a more abstract example of a two-dimensional
manifold. A manifold that looks locally Euclidean may not look
like Euclidean space globally: it may not be simply connected.
The surface of the coffee cup is not simply connected.

An example of a coordinate function is the function that maps
points in a simply-connected open neighborhood of a sphere to the
tuple of latitude and longitude.2 If we want to talk about motion
on the Earth, we can identify the space of configurations to a
2-sphere. The map from the 2-sphere to the three-dimensional
coordinates of a point on the surface of the Earth captures the
shape of the Earth.

Two angles specify the configuration of the planar double pen-
dulum. The manifold of configurations is a torus, and each angle
requires two coordinate patches to cover it (because angles wrap
around). Each point on the torus corresponds to a configuration of
the double pendulum, specifying the position of each constituent
in space. The constraints, such as the lengths of the pendulum

1We may impose a non-Euclidean metric on the coordinates. For example,
the coordinates used in relativity have a metric structure that does not obey
the requirement that distinct points have non-zero distance between them.

2The open set for a latitude-longitude coordinate system may not include
either pole (because longitude is not defined at the poles) or on the 180◦

meridian (where the longitude is discontinuous). Other coordinate systems
are needed for these places.

Figure 0.1 The coordinate function maps points on the manifold in
the coordinate patch to a tuple of coordinates. A function f on the
manifold M can be represented in coordinates by fχ = f ◦ χ−1.

rods, are built into the map between the torus and the arrange-
ments of masses in space.

Coordinate Functions

A coordinate function χ maps points in a coordinate patch of a
manifold to a coordinate tuple:3

x = χ(m), (0.1)

where x may have a convenient tuple structure. The coordinates
are arranged as an “up structure”; the coordinates are selected
with superscripts:

xi = χi(m). (0.2)

The number of independent components of x is the dimension of
the manifold.

Assume we have two coordinate functions χ and χ′. The coor-
dinate transformation from χ′ coordinates to χ coordinates is just
the composition χ ◦ χ′−1, where χ′−1 is the functional inverse of

3In the text that follows we will use sans-serif names, such as f, v, m, to refer
to objects defined on the manifold. Objects that are defined on coordinates
(tuples of real numbers) will be named with symbols like f , v, x.

Figure 0.2 Here there are two overlapping coordinate patches that are
the domains of the two coordinate functions χ and χ′. It is possible to
represent manifold points in the overlap using either coordinate system.
The coordinate transformation from χ′ coordinates to χ coordinates is
just the composition χ ◦ χ′−1.

χ′ (see figure 0.2). We assume that the coordinate transformation
is continuous and differentiable to any degree we require.

For example, we can have both rectangular and polar coordi-
nates on a plane, and

(define R2 (rectangular 2))
(define P2 (polar/cylindrical 2))

For each of these coordinate systems we obtain the coordinate
functions and their inverses.

(define R2-chi (R2 ’->coords))
(define R2-chi-inverse (R2 ’->point))
(define P2-chi (P2 ’->coords))
(define P2-chi-inverse (P2 ’->point))

The coordinate transformations are then just compositions:

(print-expression
((compose P2-chi R2-chi-inverse)
(up ’x0 ’y0)))

(up (sqrt (+ (expt x0 2) (expt y0 2))) (atan y0 x0))

(print-expression
((compose R2-chi P2-chi-inverse)
(up ’r0 ’theta0)))

(up (* r0 (cos theta0)) (* r0 (sin theta0)))

And we can obtain the Jacobian of the transformation by taking
its derivative4

(print-expression
((D (compose R2-chi P2-chi-inverse))
(up ’r0 ’theta0)))

(down (up (cos theta0) (sin theta0))
(up (* -1 r0 (sin theta0)) (* r0 (cos theta0))))

Manifold Functions:
Let f be a function on a manifold M. This function has a coor-
dinate representation fχ with respect to the coordinate function
χ:

fχ = f ◦ χ−1. (0.3)

The value of fχ(x) is independent of coordinates:

fχ(x) = (f ◦ χ−1)(χ(m)) = f(m). (0.4)

The subscript χ may be dropped when it is unambiguous.
The manifold point m is represented in rectangular coordinates

by a pair of real numbers,

(x, y) = χ(m), (0.5)

and the manifold function f is represented in rectangular coordi-
nates by f that takes a pair of real numbers and produces a real
number

f : R2 → R

4See SICM, Appendix 8, for an introduction to tuple arithmetic and a discus-
sion of derivatives of functions with structured input or output.

f : (x, y) �→ f(x, y). (0.6)

We define our manifold function

f : M → R

f : m �→ (f ◦ χ)(m). (0.7)

We can illustrate the coordinate independence with a program.
We will show that an arbitrary manifold function f, when defined
by its coordinate representation in rectangular coordinates, has
the same behavior when applied to a manifold point independent
of whether the point is specified in rectangular or polar coordi-
nates.

First define a signature for functions that map an up structure
of two reals to a real, then define a manifold function by specifying
its behavior in rectangular coordinates:5

(define R2->R (-> (UP Real Real) Real))

(define f
(compose (literal-function ’f-rect R2->R)

R2-chi))

A typical point can be specified using each coordinate system:

(define R2-point (R2-chi-inverse (up ’x0 ’y0)))
(define P2-point (P2-chi-inverse (up ’r0 ’theta0)))

The definition of the function f is independent of the coordinate
system used to specify the point of evaluation:

(print-expression (f R2-point))
(f-rect (up x0 y0))

(print-expression (f P2-point))
(f-rect (up (* r0 (cos theta0)) (* r0 (sin theta0))))

5Alternatively, we can define the same function in a shorthand

(define f (literal-manifold-function ’f-rect R2))

And, similarly, a function defined in terms of its behavior in po-
lar coordinates makes sense when applied to a point specified in
rectangular coordinates.

(define g (literal-manifold-function ’g-polar P2))

(print-expression (g P2-point))
(g-polar (up r0 theta0))

(print-expression (g R2-point))
(g-polar

(up (sqrt (+ (expt x0 2) (expt y0 2)))
(atan y0 x0)))

To make things a bit easier, we can give names to the individual
coordinate functions associated with a coordinate system. Here
we name the coordinate functions for the R2 coordinate system x
and y and for the P2 coordinate system r and theta.

(instantiate-coordinates R2 ’(x y))
(instantiate-coordinates P2 ’(r theta))

This allows us to extract the coordinates from a point, indepen-
dent of the coordinate system used to specify the point.

(print-expression (x R2-point))
x0

(print-expression (x P2-point))
(* r0 (cos theta0))

(print-expression (r P2-point))
r0

(print-expression (r R2-point))
(sqrt (+ (expt x0 2) (expt y0 2)))

(print-expression (theta R2-point))
(atan y0 x0)

We can work with the coordinate functions in a natural manner,
defining new manifold functions in terms of them:6

6This is actually a nasty, but traditional, abuse of notation. The problem is
that cos r really means cos ◦r, or in Scheme notation (square r) really means
(compose square r). Both have been made to work in our system. This

(define h (+ (* x (square r)) (cube y)))

(print-expression (h P2-point))
(+ (* (expt r0 3) (expt (sin theta0) 3))

(* (expt r0 3) (cos theta0)))

(print-expression (h R2-point))
(+ (expt x0 3)

(* x0 (expt y0 2))
(expt y0 3))

We can’t take the derivative of a manifold function; we can only
take derivatives of functions defined with real number inputs and
outputs. But we can take a real valued function on a manifold
and compose it with the inverse of a coordinate function to obtain
a function we can differentiate:

(print-expression
((D (compose h R2-chi-inverse)) R2-point))

(down (+ (* 3 (expt x0 2)) (expt y0 2))
(+ (* 2 x0 y0) (* 3 (expt y0 2))))

Vector Fields

A vector field on a manifold is an assignment of a vector to each
point on the manifold. Probably you were taught that a vector is
an arrow with a length and a direction. But we will think about
vectors in a different way.

A function on a manifold takes on various values for different
points on the manifold. A vector field takes directional derivatives
of manifold functions: it measures how each function varies as the
argument of the function is moved in the direction specified by
the vector field. Thus a vector field is an operator on functions.

Let m be a point on a manifold, v be a vector field on the
manifold, and f be a real-valued function on the manifold. Then
v(f)(m) is the directional derivative of the function f at the point m

usually does not cause trouble, but it complicates the understanding of the
derivatives, because there is a functional dependency that is implicit rather
than explicit.

in the direction specified by v.7 Note that there is no mention here
of any coordinate system. The vector field specifies a direction
and length at each manifold point that is independent of how it
is described using any coordinate system.

The direction can be specified with respect to a coordinate sys-
tem, with coordinate function χ, by giving a coefficient function
bχ,v that maps the coordinates of a point to a tuple of coefficients
corresponding to the direction in which the derivative is to be
taken. The vector field is then

v(f)(m) = ((D(f ◦ χ−1) bχ,v) ◦ χ)(m) (0.8)
= D(f ◦ χ−1)(χ(m)) bχ,v(χ(m))

=
∑

i

∂i(f ◦ χ−1)(χ(m)) bi
χ,v(χ(m)). (0.9)

We implement the definition of a vector field (0.8) as the fol-
lowing Scheme procedure

(define (vector-field-procedure components coordinate-system)
(define (the-procedure f)
(compose (* (D (compose f (coordinate-system ’->point)))

components)
(coordinate-system ’->coords)))

the-procedure)

The vector field is an operator, like derivative, made as follows:8

(define (components->vector-field components coordinate-system)
(procedure->vector-field
(vector-field-procedure components coordinate-system)))

Given a vector field v and a chart χ we can find the coefficient
function bχ,v by applying the vector field to the chart:

v(χ)(χ−1(x)) = D(χ ◦ χ−1)(x) bχ,v(x)
= D(I)(x) bχ,v(x)

7The vector field is a function that takes a real-valued manifold function and
a manifold point and produces a number. The order of arguments is chosen
to make v(f) be a new manifold function that can be manipulated further.

8An operator is just like a procedure except that multiplication is interpreted
as composition. For example, the derivative procedure is made into an oper-
ator D so that we can say (expt D 2) and expect it to compute the second
derivative.

= bχ,v(x) (0.10)

So we can use a vector field v to measure the rate of change of
the coordinates χ in the direction specified by v at each point m
on the manifold. The result is a function bχ,v of the coordinates
of the point in the manifold. Each component of bχ,v specifies the
rate of change of a coordinate in the direction and rate specified
by v.

Because the coordinates are an up structure, the derivative
makes a down structure, and the coefficient tuple bχ,v(x) is an up
structure compatible for addition to the coordinates. Note that
for any vector field v the coefficients bχ,v(x) are different for differ-
ent coordinate functions χ. In the text that follows we will usually
drop the subscripts on b, understanding that it is dependent on
the coordinate system and the vector field.

Given a coordinate system and coefficient functions that map
coordinates to real values, we can make a vector field. For exam-
ple, a general vector field can be defined by giving components
relative to the coordinate system R2 by9

(define v
(components->vector-field
(up (literal-function ’vx (-> (UP Real Real) Real))

(literal-function ’vy (-> (UP Real Real) Real)))
R2))

When this vector field is applied to an arbitrary manifold function
we see that it gives the directional derivative of that manifold
function in the direction specified by the components vx and vy.

(print-expression
((v (literal-manifold-function ’f R2)) R2-point))

(+ (* (((partial 0) f) (up x0 y0)) (vx (up x0 y0)))
(* (((partial 1) f) (up x0 y0)) (vy (up x0 y0))))

9To make it convenient to define literal vector fields we provide a shorthand:

(define w (literal-vector-field ’v R2))

This makes a vector field with component functions named vˆ0 and vˆ1 and
names the result w.

The vector field v has a coordinate representation v:

v(f)(m) = D(f ◦ χ−1)(χ(m)) b(χ(m))
= Df(x) b(x)
= v(f)(x), (0.11)

with the definitions f = f ◦ χ−1 and x = χ(m). The function b is
the coefficient function for the vector field v. It provides a scale
factor for the component in each coordinate direction. However, v
is the coordinate representation of the vector field f in that it takes
directional derivatives of coordinate representations of manifold
functions.

Given a vector field v and a coordinate system coordsys we can
construct the coordinate representation of the vector field.10

(define (coordinatize vector-field coordsys)
(define ((v f) x)
(let ((b (compose (vector-field (coordsys ’->coords))

(coordsys ’->point))))
(* ((D f) x) (b x)))))

(make-operator v))

We can apply a coordinatized vector field to a function of coordi-
nates to get the same answer as before.

(print-expression
(((coordinatize v R2)

(literal-function ’f (-> (UP Real Real) Real)))
(up ’x0 ’y0)))

(+ (* (((partial 0) f) (up x0 y0)) (vx (up x0 y0)))
(* (((partial 1) f) (up x0 y0)) (vy (up x0 y0))))

For the manifold Rn we can let χ = I. Then x = m, f = f, and

v(f)(x) = Df(x) b(x), (0.12)

which is the usual directional derivative.

Vector Field Properties:
The vector fields on a manifold form a vector space over the reals,
with the sum and scalar multiplication defined as follows. Let u

10The make-operator procedure takes a procedure and returns an operator.

and v be vector fields and let α be a real number. Then

(u + v)(f) = u(f) + v(f)
(αu)(f) = α(u(f)). (0.13)

Vector fields have the following properties. Assume f and g are
functions on the manifold, a and b are real constants, and F is a
function on the range of f.

v(af + bg)(m) = av(f)(m) + bv(g)(m)

v(af)(m) = av(f)(m)

v(fg)(m) = v(f)(m) g(m) + f(m) v(g)(m)

v(F ◦ f)(m) = DF (f(m)) v(f)(m)

(fv)(g)(m) = f(m) v(g)(m)

Coordinate Basis Vector Fields

For an n-dimensional manifold any set of n linearly independent
vector fields form a basis in that any vector field can be expressed
as a linear combination of the basis fields. Given a coordinate
system we can construct a basis as follows: we choose the compo-
nent tuple bi(x) to be the ith unit tuple ui(x)—an up tuple with
one in the ith position and zeroes in all other positions—selecting
the partial derivative in that direction. We then define the basis
vector field Xi by

Xi(f)(m) = D(f ◦ χ−1)(χ(m)) ui(χ(m))
= ∂i(f ◦ χ−1)(χ(m)). (0.14)

In terms of Xi the vector field of equation (0.9) is

v(f)(m) =
∑

i

Xi(f)(m) bi(χ(m)). (0.15)

We can also write

v(f)(m) = X(f)(m) b(χ(m)), (0.16)

letting the tuple algebra do its job.
The basis vector field is often written

∂

∂xi
= Xi, (0.17)

to call to mind that it is an operator that computes the directional
derivative in the ith coordinate direction.

In addition to making the coordinate functions the procedure
instantiate-coordinates also makes the traditional named basis
vectors:

(instantiate-coordinates R2 ’(x y))
(instantiate-coordinates P2 ’(r theta))

Using these we can examine the application of a rectangular basis
vector to a polar coordinate function:

(print-expression ((d/dx (square r)) R2-point))
(* 2 x0)

More general functions and vectors can be made as combinations
of these simple pieces:

(print-expression
(((+ d/dx (* 2 d/dy)) (+ (square r) (* 3 x))) R2-point))

(+ 3 (* 2 x0) (* 4 y0))

Coordinate transformations:
Consider a coordinate change χ = φ ◦ χ′. So if x = χ(m), x′ =
χ′(m), then x = φ(x′), and

X(f)(m) = D(f ◦ χ−1)(χ(m))
= D(f ◦ (χ′)−1 ◦ φ−1)((φ ◦ χ′)(m))
= D(f ◦ (χ′)−1)((φ−1 ◦ φ ◦ χ′)(m)) D(φ−1)((φ ◦ χ′)(m))
= D(f ◦ (χ′)−1)(χ′(m)) D(φ−1)(χ(m))
= X′(f)(m) D(φ−1)(χ(m))
= X′(f)(m) D(χ′ ◦ χ−1)(χ(m)). (0.18)

This is the rule for the transformation of basis vector fields. The
second factor can be recognized as “∂x′/∂x”.11

The vector field does not depend on coordinates. So, from
equation (0.16), we have

v(f)(m) = X(f)(m) b(χ(m))
= X′(f)(m) b′(χ′(m)) (0.19)

Using equation (0.18), we deduce

D(φ−1)(x) b(x) = b′(x′). (0.20)

Because φ−1 ◦ φ = I, we have

D(φ−1)(x) = (Dφ(x′))−1, (0.21)

and so

b(x) = Dφ(x′) b′(x′), (0.22)

as expected.12

Integral Curves

A vector field gives a direction and rate for every point on a mani-
fold. We can start at any point and go in the direction specified by
the vector field, tracing out a parametric curve on the manifold.
This curve is an integral curve of the vector field.

More formally, let v be a vector field on the manifold M. An
integral curve γv

m : R → M of v is a parametric path on M satisfying

D(f ◦ γv
m)(t) = v(f)(γv

m(t)) = (v(f) ◦ γv
m)(t) (0.23)

γv
m(0) = m, (0.24)

11This notation helps one remember the transformation rule:

∂f

∂xi
=
�

j

∂f

∂x′j
∂x′j

∂xi
,

which is the relation in the usual Leibnitz notation. Notice that f means
something different on each side of the equation.

12For coordinate paths q and q′ related by q(t) = φ(q′(t)) the velocities are re-
lated by Dq(t) = Dφ(q′(t))Dq′(t). Abstracting off paths, we get v = Dφ(x′)v′.

for arbitrary functions f on the manifold, with real or structured
real values. The rate of change of a function along an integral
curve is the vector field applied to the function evaluated at the
appropriate place along the curve. Often we will simply write γ,
rather than γv

m.
We can recover the differential equations satisfied by a coor-

dinate representation of the integral curve by letting f = χ, the
coordinate function, and letting σ = χ ◦ γ be the coordinate path
corresponding to the curve γ. Then the derivative of the coordi-
nate path σ is

Dσ(t) = D(χ ◦ γ)(t)
= (v(χ) ◦ γ)(t)
= (v(χ) ◦ χ−1 ◦ χ ◦ γ)(t)
= (b ◦ σ)(t), (0.25)

where b = v(χ) ◦ χ−1 is the coefficient function for the vector
field v for coordinates χ. So the coordinate path σ satisfies the
differential equations

Dσ = b ◦ σ. (0.26)

Differential equations for the integral curve can only be ex-
pressed in a coordinate representation, because we cannot go from
one point on the manifold to another by addition of an increment.
We can only do this by adding the coordinates to an increment
of coordinates and then finding the corresponding point on the
manifold.

Iterating the process described by equation (0.23) we can com-
pute higher-order derivatives of functions along the integral curve:

D(f ◦ γ) = v(f) ◦ γ

D2(f ◦ γ) = D(v(f) ◦ γ) = v(v(f)) ◦ γ

...

Dn(f ◦ γ) = vn(f) ◦ γ (0.27)

Thus, the evolution of f ◦ γ can be written formally as a Taylor
series in the parameter

(f ◦ γ)(t)

= (f ◦ γ)(0) + t D(f ◦ γ)(0) +
1
2
t2 D2(f ◦ γ)(0) + · · ·

= (etD(f ◦ γ))(0)
= (etvf)(γ(0)). (0.28)

In particular, let f = χ, then

σ(t) = (χ ◦ γ)(t) = (etD(χ ◦ γ))(0) = (etvχ)(γ(0)), (0.29)

a Taylor series representation of the solution to the differential
equation (0.26).

For example, a vector field J that generates a rotation about
the origin is:

(define J (- (* x d/dy) (* y d/dx)))

We can exponentiate the J vector field, to generate an evolution
in a circle around the origin starting at (1, 0):

(series:for-each print-expression
(((exp (* ’a J)) R2-chi)
((R2 ’->point) (up 1 0)))
6)

(up 1 0)
(up 0 a)
(up (* -1/2 (expt a 2)) 0)
(up 0 (* -1/6 (expt a 3)))
(up (* 1/24 (expt a 4)) 0)
(up 0 (* 1/120 (expt a 5)))

These are the first six terms of the series expansion of the coordi-
nates of the position for parameter a.

This calculation can also be done, slightly differently, in coor-
dinates. We recover the differential equation for the coordinate
path σ, from the coordinate representation of v by letting f = I,

Dσ(t) = v(I)(σ(t))
= v(χ)(χ−1(σ(t)))
= (b ◦ σ)(t). (0.30)

Consider the evolution of f ◦ σ.

D(f ◦ σ) = (Df ◦ σ) Dσ

= (Df ◦ σ) (b ◦ σ)

= v(f) ◦ σ. (0.31)

This restates equation (0.23), the constraint that γ is an integral
curve, in coordinates.

Taking another derivative

D2(f ◦ σ) = D(v(f) ◦ σ)
= (D(v(f)) ◦ σ) Dσ

= (D(v(f)) ◦ σ) (b ◦ σ)
= v(v(f)) ◦ σ. (0.32)

So successive derivatives of f ◦ σ correspond to successive appli-
cations of v to f composed with σ

Dn(f ◦ σ) = vn(f) ◦ σ. (0.33)

And thus we can exponentiate the coordinate representation of
the vector field.

We can define an evolution operator Et,v

(Et,vf)(γ(0)) = (f ◦ γ)(t) = etvf(γ(0)). (0.34)

The evolution operator also has a coordinate representation:

Et,vf(γ(0)) = Et,vf(σ(0)) = etvf(σ(0)). (0.35)

But evolution is well defined whether or not coordinates are used,
because it only depends on there being an integral curve γ. Also,
evolution may be well defined for t beyond the point at which the
Taylor series converges, even when using coordinates.

We can approximate the evolution operator by summing the
series up to a given order:

(define ((((evolution order)
delta-t vector-field)
manifold-function)
manifold-point)

(series:sum
(((exp (* delta-t vector-field))

manifold-function)
manifold-point)
order))

We can evolve J from the initial point up to the parameter a, and
accumulate the first six terms as follows:

(print-expression
((((evolution 6) ’a J) R2-chi)
((R2 ’->point) (up 1 0))))

(up (+ (* -1/720 (expt a 6))
(* 1/24 (expt a 4))
(* -1/2 (expt a 2))
1)

(+ (* 1/120 (expt a 5))
(* -1/6 (expt a 3))
a))

One-form fields

One-form fields are linear functions of vector fields that produce
real-valued functions on the manifold. A one-form field is linear
in vector fields: if ω is a one-form field, v and w are vector fields,
and c is a manifold function, then

ω(v + w) = ω(v) + ω(w)

and

ω(cv) = cω(v).

We can verify these identities in three dimensions:

(define omega (literal-1form-field ’omega R3))
(define v (literal-vector-field ’v R3))
(define w (literal-vector-field ’w R3))
(define c (literal-manifold-function ’c R3))

(print-expression
((- (omega (+ v w)) (+ (omega v) (omega w)))
R3-point))

0

(print-expression
((- (omega (* c v)) (* c (omega v))) R3-point))

0

The space of one-form fields on a manifold is a vector space over
the reals. Sums and scalar products have the expected properties,

with the following definitions: if ω and θ are one-form fields, and
if α is real number, then:

(ω + θ)(v) = ω(v) + θ(v)
(αω)(v) = α(ω(v)) (0.36)

Given a coordinate function χ, we may define the basis one-form
fields X̃i by

X̃i(v)(m) = v(χi)(m) (0.37)

or collectively

X̃(v)(m) = v(χ)(m). (0.38)

The tuple of basis one-form fields X̃(v)(m) is an up structure like
that of χ.

The basis one-form fields are dual to the basis vector fields in
the following sense:13

X̃i(Xj)(m) = Xj(χi)(m) = ∂j(χi ◦ χ−1)(χ(m)) = δi
j . (0.39)

The general one-form field ω is a linear combination of basis
one-form fields:

ω(v)(m) = a(χ(m)) X̃(v)(m) =
∑

i

ai(χ(m))X̃i(v)(m), (0.40)

with coefficient tuple a(x), for x = χ(m). We can write this more
simply

ω(v) = (a ◦ χ) X̃(v), (0.41)

because everything is evaluated at m.
The coefficient tuple can be recovered from the one-form:

a(x) = ω(X)(χ−1(x)). (0.42)

This follows from the dual relationship (0.39). We can see this as
a program:14

13The Kronecker delta δi
j is one if i = j and zero otherwise.

14Again, we provide a shortcut for this construction:

(define omega
(components->1form-field
(down (literal-manifold-function ’a 0 R3)

(literal-manifold-function ’a 1 R3)
(literal-manifold-function ’a 2 R3))

R3))

(pe ((omega (down d/dx d/dy d/dz)) R3-point))
(down (a 0 (up x0 y0 z0))

(a 1 (up x0 y0 z0))
(a 2 (up x0 y0 z0)))

The coordinate basis one-form fields can be used to find the co-
efficients of vector fields on the dual coordinate vector-field basis:

X̃i(v) = v(χi) = bi ◦ χ (0.43)

or collectively,

X̃(v) = v(χ) = b ◦ χ (0.44)

A coordinate basis one-form field is often written

dxi = X̃i (0.45)

The instantiate-coordinates procedure also makes the basis
one-form fields with these traditional names inherited from the
coordinates.

The duality of the coordinate basis vector fields and the coor-
dinate basis one-form fields is illustrated:

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))
(define R3-point

((R3 ’->point) (up ’x0 ’y0 ’z0)))

(print-expression ((dx d/dy) R3-point))
0

(print-expression ((dx d/dx) R3-point))
1

(define omega (literal-1form-field ’a R3))

We can use the coordinate basis one-form fields to extract the
coefficients of the J vector field on the rectangular vector basis:

(print-expression ((dx J) R2-point))
(* -1 y0)

(print-expression ((dy J) R2-point))
x0

But we can also find the coefficients on the polar vector basis:

(print-expression ((dr J) R2-point))
0

(print-expression ((dtheta J) R2-point))
1

So J is the same as d/dtheta, as we can see by applying them
both to the general function f:

(print-expression (((- J d/dtheta) f) R2-point))
0

Differential:
An example of a one-form field is the differential df of a manifold
function f, defined as follows. If df is applied to a vector field v
we obtain

df(v) = v(f), (0.46)

which is a function of a manifold point. The differential of a
function is linear in the vector fields. The differential is also a
linear operator on functions: if f1 and f2 are manifold functions,
and if c is a real constant, then

d(f1 + f2) = df1 + df2

and

d(cf) = cdf.

The traditional notation for the coordinate basis one-form fields
is justified by the relation:

dxi = X̃i = d(χi) (0.47)

All one-form fields can be constructed as linear combinations of
basis one-form fields, but not all one-form fields are differentials
of functions. This is why we started with the basis one-form fields
and built the general one-form fields in terms of them.

Coordinate transformations:
Under a coordinate change χ = φ ◦ χ′

X̃(v) = v(φ ◦ χ′)
= (Dφ ◦ χ′) v(χ′)

= (Dφ ◦ χ′) X̃′(v), (0.48)

where the second line follows from the chain rule for vector fields.
One-form fields are independent of coordinates. So,

ω(v) = (a ◦ χ) X̃(v)

= (a′ ◦ χ′) X̃′(v). (0.49)

With equation (0.48) this is

a(χ(m)) Dφ(χ′(m)) = a′(χ′(m)), (0.50)

or

a(χ(m)) = a′(χ′(m)) (Dφ(χ′(m)))−1. (0.51)

The coefficient tuple a(x) is a down structure compatible for
contraction with b(x). Let v be the vector with coefficient tuple
b(x), and ω be the one-form with coefficient tuple a(x). Then, by
equation (0.41),

ω(v) = (a ◦ χ) (b ◦ χ). (0.52)

As a program:

(print-expression
((omega (literal-vector-field ’v R3)) R3-point))

(+ (* (a 0 (up x0 y0 z0)) (vˆ0 (up x0 y0 z0)))
(* (a 1 (up x0 y0 z0)) (vˆ1 (up x0 y0 z0)))
(* (a 2 (up x0 y0 z0)) (vˆ2 (up x0 y0 z0))))

Comparing equation (0.51) with equation (0.22) we see that
one-form components and vector components transform oppo-
sitely, so that

a(x) b(x) = a′(x′) b′(x′). (0.53)

This shows that ω(v)(m) is independent of coordinates.

Basis Fields

A vector field may be written as a linear combination of basis
vector fields. If n is the dimension, then any set of n linearly
independent vector fields may be used as a basis. The coordinate
basis X is an example of a basis.15

Let e be a tuple of basis vector fields, such as the coordinate
basis X. The general vector field v applied to an arbitrary manifold
function f can be expressed as a linear combination

v(f)(m) = e(f)(m) b(m) =
∑

i

ei(f)(m) bi(m), (0.54)

where b is a tuple-valued coefficient function on the manifold.16

When expressed in a coordinate basis, the coefficients that specify
the direction of the vector are naturally expressed as functions
bi of the coordinates of the manifold point. Here, the coefficient
function b is more naturally expressed as a tuple-valued function
on the manifold.

The coordinate basis forms have a simple definition in terms of
the coordinate basis vectors and the coordinates (equation 0.38).
With this choice, the dual property, equation (0.39), holds without
further fuss. More generally, we can define a basis of one-forms ẽ
that is dual to e in that they satisfy the property

ẽi(ej)(m) = δi
j , (0.55)

analogous to property (0.39).

15We cannot say if the basis vectors are orthogonal or normalized until we
introduce a metric.

16If b is the coefficient function expressed as a function of coordinates, then
b = b ◦ χ is the coefficient function as a function on the manifold.

e1

e0

Figure 0.3 Let arrows e0 and e1 depict the vectors of a basis vector
field at a particular point. Then the foliations shown by the parallel
lines depict the dual basis one-form fields at that point. The dotted
lines represent the field ẽ0 and the dashed lines represent the field ẽ1.
The spacings of the lines are 1/3 unit. That the vectors pierce three
of the lines representing their duals and do not pierce any of the lines
representing the other basis elements is one way to see the relationship
ẽi(ej)(m) = δi

j .

To solve for the dual basis ẽ given the basis e express the basis
vectors e in terms of a coordinate basis

ej(f) =
∑

k

Xk(f) ck
j (0.56)

and the dual one-forms ẽ in terms of the dual coordinate one-forms

ẽi(v) =
∑

l

di
l X̃

l(v), (0.57)

then

ẽi(ej) =
∑

l

di
lX̃

l(ej)

=
∑

l

di
lej(χl)

=
∑

l

di
l

∑
k

Xk(χl)ck
j

=
∑
kl

di
lδ

l
kc

k
j

=
∑

k

di
kc

k
j . (0.58)

Applying this at m we get

ẽi(ej)(m) = δi
j =

∑
k

di
k(m)ck

j (m). (0.59)

So the d coefficients can be determined from the c coefficents (es-
sentially by matrix inversion).

The dual form fields can be used to determine the coefficients b
of a vector field v relative to a basis e, by applying the dual basis
form fields ẽ to the vector field. Let

v(f) =
∑

i

ei(f) bi. (0.60)

Then

ẽj(v) = bj . (0.61)

Define two general vector fields:

(define e0
(+ (* (literal-manifold-function ’e0x R2) d/dx)

(* (literal-manifold-function ’e0y R2) d/dy)))

(define e1
(+ (* (literal-manifold-function ’e1x R2) d/dx)

(* (literal-manifold-function ’e1y R2) d/dy)))

We use these as a vector basis and compute the dual:

(define e-vector-basis (down e0 e1))
(define e-dual-basis

(vector-basis->dual e-vector-basis P2))

Then we can verify that they satisfy the dual relationship (equa-
tion 0.39) by applying the dual basis to the vector basis:

(print-expression ((e-dual-basis e-vector-basis) m))
(up (down 1 0) (down 0 1))

Note that the dual basis was computed relative to the polar coor-
dinate system, showing how the resulting objects are independent
of the coordinates in which they were expressed!

Or we can make a general vector field with this basis and then
pick out the coefficients by applying the dual basis:

(define v
(* e-vector-basis

(up (literal-manifold-function ’bx R2)
(literal-manifold-function ’by R2))))

(print-expression ((e-dual-basis v) R2-point))
(up (bx (up x0 y0)) (by (up x0 y0)))

Commutators

The commutator of two vector fields is defined as

[v,w](f) = v(w(f)) − w(v(f)) (0.62)

The commutator of two coordinate basis fields is zero:

[Xi,Xj ](f) = Xi(Xj(f)) − Xj(Xi(f))
= ∂i∂j(f ◦ χ−1) ◦ χ − ∂j∂i(f ◦ χ−1) ◦ χ

= 0, (0.63)

because the individual partial derivatives commute.
The commutator of two vector fields is a vector field. Let v be

a vector field with coefficient function c = c ◦χ, and u be a vector
field with coefficient function b = b ◦ χ, both with respect to the
coordinate basis X. Then

[u, v](f) = u(v(f)) − v(u(f))

= u(
∑

i

Xi(f)ci) − v(
∑

i

Xi(f)bi)

=
∑

j

Xj(
∑

i

Xi(f)ci)bj −
∑

j

Xj(
∑

i

Xi(f)bi)cj

=
∑
ij

[Xi,Xj ](f)cibj

+
∑

i

Xi(f)
∑

j

(Xj(ci)bj − Xj(bi)cj)

=
∑

i

Xi(f)ai, (0.64)

where the coefficient function a of the commutator vector field is

ai =
∑

j

(
Xj(ci)bj − Xj(bi)cj

)
= u(ci) − v(bi). (0.65)

We used the fact, shown above, that the commutator of two co-
ordinate basis fields is zero.

We can check this formula for the commutator for the general
vector fields e0 and e1 in polar coordinates:

(let ((polar-vector-basis (basis->vector-basis polar-basis))
(polar-dual-basis (basis->1form-basis polar-basis)))

(print-expression
((- ((commutator e0 e1) f)

(* (- (e0 (polar-dual-basis e1))
(e1 (polar-dual-basis e0)))

(polar-vector-basis f)))
R2-point)))

0

Let e be a tuple of basis vector fields. The commutator of two
basis fields can be expressed in terms of the basis vector fields:

[ei, ej ](f) =
∑

k

dk
ijek(f), (0.66)

where dk
ij are functions of m, called the structure constants for the

basis vector fields. The coefficients are

dk
ij = ẽk([ei, ej ]). (0.67)

Define the vector fields Jx, Jy, and Jz that generate rotations
about the three rectangular axes in three dimensions:17

(define Jz (- (* x d/dy) (* y d/dx)))
(define Jx (- (* y d/dz) (* z d/dy)))
(define Jy (- (* z d/dx) (* x d/dz)))

(print-expression (((+ (commutator Jx Jy) Jz) g) R3-point))
0
(print-expression (((+ (commutator Jy Jz) Jx) g) R3-point))
0
(print-expression (((+ (commutator Jz Jx) Jy) g) R3-point))
0

We see that

[Jx, Jy] = −Jz

[Jy, Jz ] = −Jx

[Jz, Jx] = −Jy. (0.68)

You can tell if a set of basis vector fields is a coordinate basis by
calculating the commutators. If they are non-zero, then the basis
is not a coordinate basis. If they are zero then the basis vector
fields can be integrated to give the coordinate system.

Take a point 0 in M as the origin. Then, presuming [ei, ej ] = 0,
the coordinates x of the point m in the coordinate system corre-
sponding to the e basis satisfy18

m = Ex,e(0) = χ−1(x), (0.69)

where χ is the coordinate function being defined. Because the
elements of e commute, the terms in the exponential can be simply
factored into separate exponentials if needed.

17Using

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))
(define R3->R (-> (UP Real Real Real) Real))
(define g

(compose (literal-function ’g R3->R)
(R3 ’->coords)))

(define R3-point ((R3 ’->point) (up ’x ’y ’z)))

18Here x is an up-tuple structure of components, and e is down-tuple structure
of basis vectors. The product of the two contracts to make a scaled vector.

Higher-Rank Forms

A one-form field applied to a vector field is a function on the
manifold. We can extend this idea to k-form fields. A k-form field
is an antisymmetric multilinear function that takes k vector fields
and produces a function on the manifold.

Given two one-form fields ω and τ we can form a two-form field
ω ∧ τ as follows:

(ω ∧ τ)(v,w) = ω(v)τ(w) − ω(w)τ(v) (0.70)

In the special case where the one forms are the coordinate basis
forms in rectangular coordinates the wedge of them measures the
oriented area enclosed by a parallogram whose sides are described
by the vectors given as inputs:

(define R2 (rectangular 2))
(instantiate-coordinates R2 ’(x y))
(define R2-point ((R2 ’->point) (up ’x0 ’y0)))

(define v (+ (* ’a d/dx) (* ’b d/dy)))

(define w (+ (* ’c d/dx) (* ’d d/dy)))

(print-expression (((wedge dx dy) v w) R2-point))
(+ (* a d) (* -1 b c))

More generally we can form the wedge of higher-rank forms.
Let ω be a k-form field and τ be an l-form field. We can form a
k + l-form field ω ∧ τ as follows:

ω ∧ τ =
(k + l)!

k! l!
Alt(ω ⊗ τ) (0.71)

where, if η is a m-form then

Alt(η)(v0, . . . , vm−1)

=
1
m!

∑
σ∈Perm(m)

Parity(σ)η(vσ(0) , . . . , vσ(m−1)) (0.72)

and where

ω ⊗ τ(v0, . . . , vk−1, vk, . . . , vk+l−1)
= ω(v0, . . . , vk−1)τ(vk, . . . , vk+l−1). (0.73)

The wedge product is associative, and thus we need not specify
the order of a multiple application.

The 3-form that is the wedge product of the rectangular coordinate-
basis forms measures volumes:

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))
(define R3-point ((R3 ’->point) (up ’x0 ’y0 ’z0)))

(define u (+ (* ’a d/dx) (* ’b d/dy) (* ’c d/dz)))
(define v (+ (* ’d d/dx) (* ’e d/dy) (* ’f d/dz)))
(define w (+ (* ’g d/dx) (* ’h d/dy) (* ’i d/dz)))

(print-expression
(((wedge dx dy dz) u v w) R3-point))

(+ (* a e i)
(* -1 a f h)
(* -1 b d i)
(* b f g)
(* c d h)
(* -1 c e g))

This last expression is, if you don’t recognize it, the determinant
of a 3 × 3 matrix:

(print-expression
(- (((wedge dx dy dz) u v w) R3-point)

(determinant
(matrix-by-rows (list ’a ’b ’c)

(list ’d ’e ’f)
(list ’g ’h ’i)))))

0

In general, if the rank of a form is greater than the dimension
of the manifold then the form is identically zero.

Exterior Derivative

The intention of introducing the exterior derivative is to unify
all of the classical theorems of “vector analysis” into one unified
Stokes’s Theorem, which asserts that the integral of a form on the

boundary of a manifold is the integral of the exterior derivative of
the form on the interior of the manifold:∫

∂M
ω =

∫
M

dω (0.74)

In order for this to work in a way independent of the coordinate
system, the exterior derivative has to incorporate a Jacobian de-
terminant to scale the line elements, area elements, or volume
elements.

As we have seen in equation (0.46), the differential of a function
on a manifold is a one-form field. If a function on a manifold is
considered to be a form field of rank zero19 then the differential
operator increases the rank of the form by one. We can generalize
this to k-form fields with the exterior derivative operation.

Consider a one-form ω. We define

dω(v1, v2) = v1(ω(v2)) − v2(ω(v1)) − ω([v1, v2]). (0.75)

The exterior derivative of a k-form field is a k + 1-form field,
given by:20

dω(v0, . . . , vk)

=
k∑

i=0

((−1)ivi(ω(v0, . . . , vi−1, vi+1, . . . , vk)) +

k∑
j=i+1

(−1)i+jω([vi, vj ], v0, . . . , vi−1, vi+1, . . . , vj−1, vj+1, . . . , vk))

(0.76)

This formula is coordinate-system independent. This is the way
we compute the exterior derivative in our software.

19A manifold function f induces a form field f̂ of rank 0 as follows:

f̂()(m) = f(m).

20See Spivak, Differential Geometry, Volume 1, p.289.

If the form field ω is represented in a coordinate basis

ω =
n−1∑

i0=0,...,ik−1=0

ωi0,...,ik−1dxi0 ∧ · · · ∧ dxik−1 (0.77)

then the exterior derivative can be expressed as

dω =
n−1∑

i0=0,...,ik−1=0

dωi0,...,ik−1 ∧ dxi0 ∧ · · · ∧ dxik−1 . (0.78)

Though this formula is expressed in terms of a coordinate basis,
the result is independent of the choice of coordinate system.

We can test that the computation indicated by equation (0.76)
is equivalent to the explicit computation indicated by equation (0.78)
in three dimensions

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))
(define R3-point ((R3 ’->point) (up ’x0 ’y0 ’z0)))

with a general one-form field:

(define a (literal-manifold-function ’alpha R3))
(define b (literal-manifold-function ’beta R3))
(define c (literal-manifold-function ’gamma R3))

(define theta (+ (* a dx) (* b dy) (* c dz)))

The test will require two arbitrary vector fields

(define X (literal-vector-field ’X R3))
(define Y (literal-vector-field ’Y R3))

(print-expression
(((- (d theta)

(+ (wedge (d a) dx)
(wedge (d b) dy)
(wedge (d c) dz)))

X Y)
R3-point))

0

We can also try a general 2-form field in three-dimensional space:
Let

ω = ady ∧ dz + bdz ∧ dx + cdx ∧ dy, (0.79)

where a = α ◦ χ, b = β ◦ χ, c = γ ◦ χ, and α, β, and γ are
real-valued functions of three real arguments. As a program,

(define omega
(+ (* a (wedge dy dz))

(* b (wedge dz dx))
(* c (wedge dx dy))))

Here we need another vector field because our result will be a
3-form field.

(define Z (literal-vector-field ’Z R3))

(print-expression
(((- (d omega)

(+ (wedge (d a) dy dz)
(wedge (d b) dz dx)
(wedge (d c) dx dy)))

X Y Z)
R3-point))

0

A form field ω that is the exterior derivative of another form
field ω = dθ is called exact. A form field whose exterior derivative
is zero is called closed.

Every exact form field is a closed form field, i.e. applying the
exterior derivative operator twice always yields zero:

d2ω = 0. (0.80)

This is equivalent to the statement that partial derivatives with
respect to different variables commute.21

Consider the general one-form field θ defined on 3-dimensional
rectangular space. Two exterior derivatives of θ yields a 3-form
field, which must be proportional to the volume element.22 It
turns out to be zero:

21See Spivak, Calculus on Manifolds, p.92

22The dimension of the space of k-forms in an n-dimensional space is n−k+1.

(print-expression
(((d (d theta)) X Y Z) R3-point))

0

Relationship to Vector Calculus:
In three-dimensional Euclidean space the traditional vector deriva-
tive operations are gradient, curl, and divergence. If x̂, ŷ, ẑ are the
usual orthonormal vector basis, f a function on the space, and �v
is a vector field on the space, then

grad(f) =
∂f

∂x
x̂ +

∂f

∂y
ŷ +

∂f

∂z
ẑ

curl(�v) =
(

∂vz

∂y
− ∂vy

∂z

)
x̂ +

(∂vx

∂z
− ∂vz

∂x

)
ŷ +

(
∂vy

∂x
− ∂vx

∂y

)
x̂

div(�v) =
∂vx

∂x
+

∂vy

∂y
+

∂vz

∂z

These vector calculus operations are subsumed by exterior deriva-
tives of form fields, as follows. Let θ be a one-form field and let ω
be a two-form field:

θ = θxdx + θydy + θzdz

ω = ωxdy ∧ dz + ωydz ∧ dx + ωzdx ∧ dy

The exterior-derivative expressions corresponding to the vector-
calculus expressions are:

df =
∂f

∂x
dx +

∂f

∂y
dy +

∂f

∂z
dz

dθ =
(

∂θz

∂y
− ∂θy

∂z

)
dy ∧ dz +

(∂θx

∂z
− ∂θz

∂x

)
dz ∧ dx

+
(

∂θy

∂x
− ∂θx

∂y

)
dx ∧ dy

dω =
(

∂ωx

∂x
+

∂ωy

∂y
+

∂ωz

∂z

)
dx ∧ dy ∧ dz.

Vector Integral Theorems:
Green’s Theorem states that for an arbitrary compact set M ⊂ R2∫

∂M
(α ◦ χ) dx + (β ◦ χ) dy =

∫
M

((∂0β − ∂1α) ◦ χ) dx ∧ dy (0.81)

We can test this. By Stokes’s theorem, the integrands are related
by an exterior derivative. First we need a plane.

(define R2 (rectangular 2))
(instantiate-coordinates R2 ’(x y))
(define R2-chi (R2 ’->coords))
(define R2->R (-> (UP Real Real) Real))

We also need some vectors to test our forms:

(define v (literal-vector-field ’v R2))
(define w (literal-vector-field ’w R2))

We can now test our integrands:

(define alpha (literal-function ’alpha R2->R))
(define beta (literal-function ’beta R2->R))

(print-expression
(((- (d (+ (* (compose alpha R2-chi) dx)

(* (compose beta R2-chi) dy)))
(* (compose (- ((partial 0) beta)

((partial 1) alpha))
R2-chi)

(wedge dx dy)))
v w)
R2-point))

0

We can also compute the integrands for the Divergence Theo-
rem: For an arbitrary compact set M ⊂ R3 and a vector field w∫

M
div(w)dV =

∫
∂M

w · n dA (0.82)

where n is the outward pointing normal to the surface ∂M . Again,
the integrands should be related by an exterior derivative, if this
is an instance of Stokes’s Theorem.

Let

w = a
∂

∂x
+ b

∂

∂y
+ c

∂

∂z
(0.83)

We interpret w · ndA as the two-form

ω = ady ∧ dz + bdz ∧ dx + cdx ∧ dy, (0.84)

and div(w)dV as the three-form

dω =
(

∂

∂x
a +

∂

∂y
b +

∂

∂z
c

)
dx ∧ dy ∧ dz. (0.85)

Let’s compute this to make sure it is true:

(define domega
(* (+ (d/dx a) (d/dy b) (d/dz c))

(wedge dx dy dz)))

(print-expression
(((- (d omega) domega) X Y Z) R3-point))

0

Over a Map

To deal with motion on manifolds we need to think about paths
on manifolds and vectors along these paths. Vectors along paths
are not vector fields on the manifold because they are only defined
on the path. And the path may even cross itself, which would give
more than one vector at a point. Here we introduce the concept
of over a map, which solves this problem.23

Let μ map points n in the manifold N to points m in the man-
ifold M. A vector over the map μ takes directional derivatives
of functions on M at points m = μ(n). The vector over the map
applied to the function on M is a function on N.

Let v be a vector field on M, and f a function on M. Then

vμ(f) = v(f) ◦ μ, (0.86)

is a vector over the map μ. Note that vμ(f) is a function on N,
not M:

vμ(f)(n) = v(f)(μ(n)). (0.87)

We can implement this definition as:

23see Bishop and Goldberg, Tensor Analysis on Manifolds.

(define ((vector-field-over-map mu:N->M) v-on-M)
(procedure->vector-field
(lambda (f-on-M)
(compose (v-on-M f-on-M) mu:N->M))))

Given a one-form ω, the one-form over the map μ is constructed
as follows:

ωμ(vμ)(n) = ω(u)(μ(n)), where u(f)(m) = vμ(f)(n). (0.88)

The object u is not really a vector field on M even though we have
given it that shape so that the dual vector can apply to it; u(f) is
only evaluated at images m = μ(n) of points n in N. If we were
defining u as a vector field we would need the inverse of μ to find
the point n = μ−1(m), but this is not required to define the object
u in a context where there is already an m associated with the n
of interest. To extend this idea to k-forms, we carry each vector
argument over the map.

The procedure that implements u, that is, that makes a vector
field over the map μ : N → M appear as a vector field on M is

(define (vector-field-over-map->vector-field V-over-mu n)
(procedure->vector-field
(lambda (f)
(lambda (m) ((V-over-mu f) n)))))

Using this, the procedure that constructs a k-form over the map
from a k-form is:

(define ((form-field-over-map mu:N->M) w-on-M)
(let ((k (get-rank w-on-M)))
(procedure->nform-field
(lambda vectors-over-map

(lambda (n)
((apply w-on-M

(map (lambda (V-over-mu)
(vector-field-over-map->vector-field
V-over-mu n))

vectors-over-map))
(mu:N->M n))))

k)))

Let e be a tuple of basis vector fields, and ẽ be the tuple of basis
one-forms that is dual to e:

ẽi(ej)(m) = δi
j . (0.89)

The basis vectors over the map, eμ, are particular cases of vectors
over a map:

eμ(f) = e(f) ◦ μ. (0.90)

And the elements of the dual basis over the map, ẽμ, are particular
cases of one-forms over the map. The basis and dual basis over
the map satisfy

ẽi
μ(eμ

j )(n) = δi
j . (0.91)

For example, let μ map the time line to the unit sphere. We
use colatitude θ and longitude φ as coordinates on the sphere

(define sphere (S2 1)) ; sphere with R=1
(instantiate-coordinates sphere ’(theta phi))
(define sphere-basis (coordinate-system->basis sphere))

and let t be the coordinate of the real line.

(instantiate-coordinates the-real-line ’t)

A general path on the sphere is:

(define mu
(compose (sphere ’->point)

(up (literal-function ’alpha)
(literal-function ’beta))

(the-real-line ’->coords)))

The basis over the map is constructed from the basis on the sphere:

(define sphere-basis-over-mu
(basis->basis-over-map mu sphere-basis))

(define h
(compose (literal-function ’h R2->R)

(sphere ’->coords)))

(print-expression
(((basis->vector-basis sphere-basis-over-mu) h)
((the-real-line ’->point) ’t0)))

(down
(((partial 0) h) (up (alpha t0) (beta t0)))
(((partial 1) h) (up (alpha t0) (beta t0))))

The basis vectors over the map compute derivatives of the function
h evaluated on the path at the given time.

We can check that the dual basis over the map does the correct
thing

(print-expression
(((basis->1form-basis sphere-basis-over-mu)
(basis->vector-basis sphere-basis-over-mu))
((the-real-line ’->point) ’t0)))

(up (down 1 0) (down 0 1))

Pullback of a Function:
The pullback of a function f on M over the map μ is defined as

μ∗f = f ◦ μ. (0.92)

This allows us to take a function defined on M and use it to define
a new function on N.

For example, the integral curve of v evolved for time t as a
function of the initial manifold point m generates a map φv

t of the
manifold onto itself. This is a simple Currying of the integral curve
of v from m as a function of time: φv

t (m) = γv
m(t). The evolution

of the function f along an integral curve, equation (0.34), can be
written in terms of the pullback over φv

t :

(Et,vf)(m) = f(φv
t (m)) = ((φv

t )
∗f)(m). (0.93)

Pushforward of a Vector Field:
We can also pushforward a vector field over the map μ. The push-
forward takes a vector field v defined on N. The result takes
directional derivatives of functions on M at a place determined by
a point in M:

μ∗v(f)(m) = v(μ∗f)(μ−1(m)) = v(f ◦ μ)(μ−1(m)). (0.94)

We can pushforward a vector field over the map generated by an
integral curve, because the inverse is always available.24

((φv
t )∗v)(f)(m) = v((φv

t )
∗f)(φv

−t(m)) = v(f ◦ φv
t )(φ

v
−t(m)). (0.95)

Differential of a Map:
For more general maps the pushforward of a vector field is not
very useful because it requires inverting the map μ. A more useful
construct in this context is the differential of the map

dμ(v)(f)(n) = v(μ∗f)(n) = v(f ◦ μ)(n), (0.96)

which takes its argument in the source manifold N. The differen-
tial of a map applied to a vector yields a vector over the map. A
procedure to compute the differential is:

(define (((differential mu) v) f)
(v (compose f mu)))

The nomenclature of this subject is confused. The “differential
of a map” between manifolds dμ takes one more argument than
the “differential of a real-valued function on a manifold” df, but
the two are related:

dμ(v)(I)(n) = dμ(v)(n), (0.97)

where the target manifold of μ is the reals and I is the identity
function on the reals.

Let μ be the map from the time line to the manifold M, and ∂/∂t
be a basis vector on the time line. Then dμ(∂/∂t) is the vector
over the map μ that computes the rate of change of functions on
M along the path μ. This is the velocity vector.

Let χ be a tuple of coordinates on M, with associated basis
vectors Xi, and dual basis elements dxi. The vector basis and
dual basis over the map μ are Xμ

i and dxi
μ. The components of

the velocity (rates of change of coordinates along the path μ) are
obtained by applying the dual basis over the map to the velocity

vi(t) = dxi
μ(dμ(∂/∂t))(t), (0.98)

24The map φv
t is always invertible: (φv

t)
−1 = φv

−t because of the uniqueness of
the solutions of the initial value problem for ordinary differential equations.

where t is the coordinate for the point t.
For example, the velocities on a sphere are

(print-expression
(((basis->1form-basis sphere-basis-over-mu)
((differential mu) d/dt))
((the-real-line ’->point) ’t0)))

(up ((D alpha) t0) ((D beta) t0)))

as expected.

Pullback of a Vector Field:
Given a vector field v on manifold M we can pullback the vector
field through the map μ : N → M as follows:

μ∗v = (μ−1)∗v, (0.99)

or

μ∗v(f) = (v(f ◦ μ−1)) ◦ μ (0.100)

This may be useful when the map is invertible, as in the flow
generated by a vector field.

Pullback of a Form Field:
We can also pullback a one-form field ω defined on M, but an
honest definition is rarely written. The pullback of a one-form
field applied to a vector field is intended to be the same as the
one-form field applied to the pushforward of the vector field.

The pullback of a one-form field is often described by the rela-
tion

μ∗(ω)(v) = ω(μ∗(v)), (0.101)

but this is wrong, because the two sides are not functions of points
in the same manifold. The one-form field ω must apply to a vector
field on the manifold M, which takes a directional derivative of a
function g defined on M and is evaluated at a place m on M, but
the left-hand side is evaluated at a place in the manifold N.

A more precise description would be

μ∗(ω)(v)(n) = ω(μ∗(v))(μ(n)). (0.102)

Although this is accurate, it may not be effective, because com-
puting the pushforward requires the inverse of the map μ. Again,
this is available when the map is the flow generated by a vector
field.

In fact it is possible to compute the pullback of a one-form
field without having the inverse of the map. Using the Church
λ-calculus we can define a function that looks like a vector field
on M, but which evaluates the vector field over the map at a point
in N:

C(vμ, n) = λ(g)(λ(m)vμ(g)(n)). (0.103)

Note that the bound variable m is ignored and there are no in-
verses. This is cumbersome in traditional mathematical notation,
because of the unusual routing of arguments. But, unwinding the
λ-expressions:

C(vμ, n)(g)(m) = vμ(g)(n). (0.104)

Using this idea we can write the pullback of a one-form field as
follows:

μ∗(ω)(v)(n) = ω(C(dμ(v), n))(μ(n)) (0.105)

We can show that the definition (0.105) implements the require-
ment (0.102). This only requires showing

μ∗(v) = C(dμ(v), n), (0.106)

at m = μ(n). The latter condition is always satisfied in the con-
text (0.105).

The pullback of a k-form generalizes equation 0.102:

μ∗(ω)(u, v, . . .)(n) = ω(μ∗(u), μ∗(v), . . .)(μ(n)). (0.107)

Properties of Pullback:
The pullback through a map has many nice properties: it dis-
tributes through addition and through wedge product:

μ∗(θ + φ) = μ∗θ + μ∗φ (0.108)
μ∗(θ ∧ φ) = μ∗θ ∧ μ∗φ (0.109)

The pullback also commutes with the exterior derivative:

d(μ∗(θ)) = μ∗(d(θ)), (0.110)

for θ a function or k-form field.
We can verify this by computing an example. Let μ map the

rectangular plane to rectangular 3-space:25

(define mu
(compose R3-chi-inverse

(up (literal-function ’mux R2->R)
(literal-function ’muy R2->R)
(literal-function ’muz R2->R))

R2-chi))

First, let’s compare the pullback of the exterior derivative of a
function with the exterior derivative of the pullback of the func-
tion.

(define f (compose (literal-function ’f R3->R) R3-chi))

(print-expression
(((- ((pullback mu) (d f)) (d ((pullback mu) f)))

X2)
R2-point))

0

More generally, we can consider what happens to a form field. For
a one-form field the result is as expected.

25We use the following definitions:

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))
(define R3-chi (R3 ’->coords))
(define R3-chi-inverse (R3 ’->point))
(define R3->R (-> (UP Real Real Real) Real))
(define R3-point (R3-chi-inverse (up ’x0 ’y0 ’z0)))

(define R2 (rectangular 2))
(instantiate-coordinates R2 ’(u v))
(define R2-chi (R2 ’->coords))
(define R2-chi-inverse (R2 ’->point))
(define R2->R (-> (UP Real Real) Real))
(define R2-point (R2-chi-inverse (up ’u0 ’v0)))
(define X2 (literal-vector-field ’X R2))
(define Y2 (literal-vector-field ’Y R2))

(define theta (literal-1form-field ’theta R3))

(print-expression
(((- ((pullback mu) (d theta)) (d ((pullback mu) theta)))

X2 Y2)
R2-point))

0

Lie Derivative

The Lie derivative is a measure of how things change along the
integral curves of a vector field. The Lie derivative of any object
is another object of the same kind. The form of the Lie derivative
depends on the kind of object that is being differentiated. How-
ever, we will see that there are uniform ways of specifying the Lie
derivative.26

Functions:
The simplest kind of object is a function on the manifold. Let
f be a function on the manifold, and φv

t (m) be the point along
the integral curve of the vector field v beginning at m, advanced
for interval t. Then the Lie derivative of f with respect to v is a
function defined by:

Lv(f)(m) = lim
t→0

(
((φv

t )
∗f)(m) − f(m)

t

)
(0.111)

= lim
t→0

(
f(φv

t (m)) − f(m)
t

)
= v(f)(m) (0.112)

So the Lie derivative of a function is just the derivative of the
function along the integral curve of v.

Vector Fields:
The Lie derivative of a vector field y with respect to a vector field
v is a vector field that is defined by its behavior when applied
to an arbitrary manifold function f. The Lie derivative compares

26Thanks to Will Farr, who showed us a unified framework for understanding
Lie derivatives.

the pullback of the vector field y along the integral curves of the
vector field v with the original vector field y:

Lvy(f)(m) = lim
t→0

(
((φv

t )∗y)(f)(m) − y(f)(m)
t

)
(0.113)

= lim
t→0

(
((φv−t)∗y)(f)(m) − y(f)(m)

t

)
(0.114)

= lim
t→0

(
(φv

t )∗(y((φv−t)∗f))(m) − y(f)(m)
t

)
(0.115)

= lim
t→0

(
y((φv−t)

∗f)(m) + tv(y((φv−t)
∗f))(m) − y(f)(m)

t

)
= lim

t→0
(v(y((φv

−t)
∗f))(m)) − y(v(f))(m)

= v(y(f))(m) − y(v(f))(m) (0.116)
= [v, y](f)(m) (0.117)

So the Lie derivative of one vector field with respect to another
is just their commutator. Note that the tricky step (0.115) in the
derivation above depends on Hadamard’s Lemma,27 which allows
us to deduce that: (φv

t )
∗g = g ◦ φv

t = g + tv(g) + O(t2). We apply
this in equation (0.115), defining g = y((φv−t)

∗f).
The Lie derivative of a vector field can also be seen as a com-

parison of the advance of the application of the vector field to the
function with the application of the vector field to the advance of
the function:

Lvy(f)(m) = lim
t→0

(
(φv

t )∗(y(f))(m) − y((φv
t )∗f)(m)

t

)
= lim

t→0

(
y(f)(φv

t (m)) − y(f ◦ φv
t )(m)

t

)
(0.118)

We can write this as a derivative28

Lvy(f)(m) = Dg(0), (0.119)

27See Theodore Frankel, The Geometry of Physics, Cambridge University
Press 1997, pp. 126–127.

28Using L’Hospital’s rule.

with

g(t) = y(f)(φv
t (m)) − y(f ◦ φv

t )(m). (0.120)

The derivative of the first term of g is of the form of the deriva-
tive of a function h = y(f) along the integral curve of v starting at
m, so this term is just v(y(f))(m).

Let h(t) = y(f ◦ φv
t )(m), the second term of g(t). We need to

calculate Dh(0). Now,29

(f ◦ φv
t ) = f + t v(f) + · · · , (0.121)

and since y is linear

y(f ◦ φv
t ) = y(f) + t y(v(f)) + · · · . (0.122)

So Dh(0) = y(v(f))(m). Combining these two terms we get the
commutator: so

Lvy(f) = [v, y](f), (0.123)

as before.

Exponentiating Lie Derivatives:
We can exponentiate the Lie derivative and apply it to a vector
field:

eLvy = y + Lvy +
1
2
Lv

2y + · · ·

= y + [v, y] +
1
2
[v, [v, y]] + · · · (0.124)

Consider a simple case. We advance the coordinate basis vector
∂/∂y by an angle a around the circle. Let J = x ∂/∂y − y ∂/∂x,
the circular vector field. We recall

(define J (- (* x d/dy) (* y d/dx)))

We can apply the exponential of Lie derivative with respect to J
to ∂/∂y. We examine how the result affects a general function on
the manifold:

29Again, by Hadamard’s Lemma.

(series:for-each print-expression
((((exp (* ’a (Lie-derivative J))) d/dy)
(literal-manifold-function ’f R2))
((R2 ’->point) (up 1 0)))
5)

(((partial 1) f) (up 1 0))
(* a (((partial 0) f) (up 1 0)))
(* -1/2 (expt a 2) (((partial 1) f) (up 1 0)))
(* -1/6 (expt a 3) (((partial 0) f) (up 1 0)))
(* 1/24 (expt a 4) (((partial 1) f) (up 1 0)))
;Value: ...

Apparently the result is

exp
(
La (x ∂/∂y−y ∂/∂x)

) ∂

∂y
= sin(a)

∂

∂x
+ cos(a)

∂

∂y
. (0.125)

Form fields:
We can also define the Lie derivative of a form field. Consider a
one-form field ω. Its Lie derivative with respect to a vector field
v is a one-form field defined by its action on an arbitrary vector
field y:

Lvω(y)(m) = lim
t→0

(
(φv

t )∗ω(y)(m) − ω(y)(m)
t

)
(0.126)

= lim
t→0

(
ω((φv

t )∗y)(φv
t (m)) − ω(y)(m)

t

)
= lim

t→0

(
v(ω(φv

t )∗y)(m) +
ω((φv

t )∗y)(m) − ω(y)(m)
t

)
= v(ω(y))(m) − ω(Lvy)(m) (0.127)

The Lie derivative of a k-form field ω with respect to a vector
field v is a k-form field that is defined by its behavior when applied
to k arbitrary vector fields w0, . . . ,wk−1. Using the fact that the
pullback of a k-form is the form applied to the pushforward of
all its arguments evaluated at the right place, we can generalize
the derivation for one-form fields to k-form fields, by the following
formula:

Lvω(w0, . . . ,wk−1)

= v(ω(w0, . . . ,wk−1)) −
k−1∑
i=0

ω(w0, . . . ,Lvwi, . . . ,wk−1)(0.128)

Uniform Interpretation
Notice that in each case, the Lie derivative can be written in
a form reminiscent of an elementary derivative. Consider equa-
tions (0.111), (0.113), and (0.126). Each is a limit of a quotient
of a difference of a pullback of a thing and the thing. If we under-
stand the pullback as the advance along the integral curve then
this is the ordinary derivative.

Another uniform way of writing the Lie derivative is the vector
field applied to the composite function minus the function applied
to the Lie derivative of its arguments. For a function, the Lie
derivative is just

Lv(f) = v(f). (0.129)

For a vector field (see equation 0.116)

Lvy(f) = v(y(f)) − y(Lv(f)). (0.130)

For a one-form field (see equation 0.127)

Lvω(y) = v(ω(y)) − ω(Lvy). (0.131)

The generalization for k-form fields is just equation (0.128).

Properties of the Lie Derivative:
For any k-form field ω and any vector field v the exterior derivative
commutes with the Lie derivative with respect to the vector field:

Lv(dω) = d(Lvω) (0.132)

We can verify this in three-dimensional rectangular space for our
general one form:

(define V (literal-vector-field ’V R3))

(print-expression
(((- ((Lie-derivative V) (d theta))

(d ((Lie-derivative V) theta)))
X Y)
R3-point))

0

and for the general 2-form:

(print-expression
(((- ((Lie-derivative V) (d omega))

(d ((Lie-derivative V) omega)))
X Y Z)
R3-point))

0

The Lie derivative satisfies a another nice elementary relation-
ship. If v and w are two vector fields then

[Lv,Lw] = L[v,w]. (0.133)

Again, for our general one-form θ:

(print-expression
((((- (commutator (Lie-derivative X) (Lie-derivative Y))

(Lie-derivative (commutator X Y)))
theta)
Z)
R3-point))

0

and for the two-form ω:

(print-expression
((((- (commutator (Lie-derivative X) (Lie-derivative Y))

(Lie-derivative (commutator X Y)))
omega)
Z V)
R3-point))

0

Interior Product

There is a simple, but useful operation available between vector
fields and form fields called interior product. This is the substitu-
tion of a vector field v into the first argument of an p-form field ω
to produce an p − 1-form field:

(ivω)(v1, . . . vp−1) = ω(v, v1, . . . vp−1) (0.134)

There is a rather nice identity for the Lie derivative in terms of
the interior product called Cartan’s formula:

Lvω = iv(dω) + d(ivω) (0.135)

We can verify Cartan’s formula in a simple case with a program:

(define R3 (rectangular 3))
(instantiate-coordinates R3 ’(x y z))

(define X (literal-vector-field ’X R3))
(define Y (literal-vector-field ’Y R3))
(define Z (literal-vector-field ’Z R3))

(define alpha (literal-manifold-function ’alpha R3))
(define beta (literal-manifold-function ’beta R3))
(define gamma (literal-manifold-function ’gamma R3))

(define omega
(+ (* alpha (wedge dx dy))

(* beta (wedge dy dz))
(* gamma (wedge dz dx))))

(define ((L1 X) omega)
(+ ((interior-product X) (d omega))

(d ((interior-product X) omega))))

(print-expression
((- (((Lie-derivative X) omega) Y Z)

(((L1 X) omega) Y Z))
((R3 ’->point) (up ’x0 ’y0 ’z0))))

0

Note that iv ◦ iu + iu ◦ iv = 0. One consequence of this is that
iv ◦ iv = 0.

Covariant Derivative of Vector Fields

Vectors at the same manifold point can be directly compared,
but vectors at different points cannot. Parallel transport allows
vectors at different points to be compared by transporting them to
the same place. To define parallel transport, and the associated
idea of covariant derivative, we need to introduce a connection.
The connection cannot be derived, and there is freedom in the

choice of a connection. If there is a metric, then there is a special
connection determined by it.

The covariant derivative with respect to the vector field u of
the vector field v is

(∇uv)(f) =
∑

i

ei(f)

(
u(ẽi(v)) +

∑
j

ωi
j(u)ẽj(v)

)

=
∑

i

ei(f)

(
u(vi) +

∑
j

ωi
j(u)vj

)
= e(f)(u(v) + ω(u)v), (0.136)

where ω is a structure-valued one-form and ωi
j is a one-form for

each selector i and j. These one-forms are called the Cartan one-
forms, or the connection one-forms. The Cartan forms are defined
with respect to the basis e and its dual ẽ. In these expressions v
is the structure of coefficient manifold functions for the vector v
with respect to the basis e, and vi is a component of v. Though
the covariant derivative is expressed with respect to a basis, the
result is independent of the basis.

As a program, the covariant derivative is:30

30This program is slightly incomplete, for two reasons. It must construct a
vector field, and it must make a differential operator. The full program is a
bit longer:

(define (covariant-derivative Cartan)
(let ((basis (Cartan->basis Cartan))

(Cartan-forms (Cartan->forms Cartan)))
(let ((vector-basis (basis->vector-basis basis))

(1form-basis (basis->1form-basis basis)))
(define (nabla X)
(define (nabla X V)

(let ((v-components (1form-basis V)))
(let ((deriv-components

(+ (X v-components)
(* (Cartan-forms X) v-components))))

(define (the-derivative f)
(* (vector-basis f) deriv-components))

(procedure->vector-field the-derivative
‘((nabla ,(diffop-name X)) ,(diffop-name V))))))

(procedure->vector-field nabla X
‘(nabla ,(diffop-name X))))

nabla)))

(define ((((covariant-derivative Cartan) X) V) f)
(let ((basis (Cartan->basis Cartan))

(Cartan-forms (Cartan->forms Cartan)))
(let ((vector-basis (basis->vector-basis basis))

(1form-basis (basis->1form-basis basis)))
(let ((v-components (1form-basis V)))
(* (vector-basis f)

(+ (X v-components)
(* (Cartan-forms X) v-components)))))))

The Cartan forms can be constructed from the dual basis one-
forms:

ωi
j(v)(m) =

∑
k

Γi
jk(m) ẽk(v)(m). (0.137)

The connection coefficient functions Γi
jk are sometimes called the

Christoffel symbols.31 Making use of the structures, the Cartan
forms are

ω(v) = Γ ẽ(v). (0.138)

Conversely, the Christoffel coefficients may be obtained from the
Cartan forms

Γi
jk = ωi

j(ek). (0.139)

The covariant derivative is independent of the coordinate sys-
tem. We can show this in a simple case, using rectangular and

31This terminology may be restricted to the case in which the basis is a coor-
dinate basis.

polar coordinates in the plane.32 The Christoffel coefficients for
rectangular coordinates in a (flat) plane are all zero:33

(define rectangular-Christoffel
(make-Christoffel
(let ((zero (lambda (m) 0)))
(down (down (up zero zero)

(up zero zero))
(down (up zero zero)

(up zero zero))))
rectangular-basis))

The Christoffel coefficients for polar coordinates on the plane
are:34

(define polar-Christoffel
(make-Christoffel
(let ((zero (lambda (m) 0)))
(down (down (up zero zero)

(up zero (/ 1 r)))
(down (up zero (/ 1 r))

(up (* -1 r) zero))))
polar-basis))

From these we can make the Cartan forms in both coordinate
systems:35

32We will need a few definitions:

(define R2 (rectangular 2))
(instantiate-coordinates R2 ’(x y))
(define rectangular-basis (coordinate-system->basis R2))
(define R2-chi (R2 ’->coords))
(define R2-chi-inverse (R2 ’->point))

(define P2 (polar/cylindrical 2))
(instantiate-coordinates P2 ’(r theta))
(define polar-basis (coordinate-system->basis P2))
(define P2-chi (P2 ’->coords))
(define P2-chi-inverse (P2 ’->point))

33Since the Christoffel coefficients are basis dependent they are packaged with
the basis.

34We will derive these later.

35The code for making the Cartan forms is as follows:

(define rectangular-Cartan
(Christoffel->Cartan rectangular-Christoffel))

(define polar-Cartan
(Christoffel->Cartan polar-Christoffel))

The vector field Jz generates a rotation in the plane (the same
as J). The covariant derivative with respect to ∂/∂x of Jz applied
to an arbitrary manifold function is:

(define f
(compose (literal-function ’f-rect R2->R) R2-chi))

(print-expression
(((((covariant-derivative rectangular-Cartan) d/dx)

J)
f)
R2-point))

(((partial 1) f-rect) (up x0 y0))

Note that this is the same thing as ∂/∂y applied to the function:

(print-expression ((d/dy f) R2-point))
(((partial 1) f-rect) (up x0 y0))

In rectangular coordinates, where the Christoffel coefficients are
zero, the covariant derivative ∇uv is the vector whose coefficents
are obtained by applying u to the coefficients of v. Here, only
one coefficient of Jz depends on x, the coefficient of ∂/∂y, and it
depends linearly on x. So ∇∂/∂xJz = ∂/∂y.

Note that we get the same answer if we use polar coordinates
to compute the covariant derivative:

(print-expression
(((((covariant-derivative polar-Cartan) d/dx)

J)
f)
R2-point))

(((partial 1) f-rect) (up x0 y0))

(define (Christoffel->Cartan Christoffel)
(let ((basis (Christoffel->basis Christoffel))

(Christoffel-symbols (Christoffel->symbols Christoffel)))
(make-Cartan
(* Christoffel-symbols (basis->1form-basis basis))
basis)))

’ν

d/dy

d/dx

ν

Δν

Figure 0.4 If v and v′ are ”arrow” representations of vectors in the
circular field and we parallel transport v to v′ in the ∂/∂x direction,
then the difference is in the ∂/∂y direction.

In rectangular coordinates the Christoffel coefficients are all zero;
in polar coordinates there are non-zero coefficients, but the value
of the covariant derivative is the same. In polar coordinates the
basis elements vary with position, and the Christoffel coefficients
compensate for this.36

Of course, this is a pretty special situation. Let’s try something
more general.

(define v (literal-vector-field ’v R2))
(define w (literal-vector-field ’w R2))

(print-expression
(((((- (covariant-derivative rectangular-Cartan)

(covariant-derivative polar-Cartan))
v)
w)
f)
R2-point))

0

36Non-zero Christoffel coefficients also appear if the manifold has curvature,
as we will see.

Parallel Transport

Given a connection we can define the operation of parallel trans-
port of a vector along a path. Let γ be a path, a map from the
real line to the manifold M. We would like to parallel transport
w, a vector over the map γ, along the path γ.

We can specify w in terms of a basis over the map eγ as follows:

w(f)(t) =
∑

i

eγ
i (f)(t)ci(t) = (eγ(f) c)(t). (0.140)

The coefficients c are functions on the time line and have struc-
tured values that are compatible for contraction with a basis on
M.

The covariant derivative is expressed with respect to a basis
and a set of Cartan forms. If we replace these with a basis over
a map and the Cartan forms over the map then we get a covari-
ant derivative over the map. The resulting covariant derivative is
independent of the particular basis chosen, but now depends on
the map, notated ∇γ . The construction of the basis over the map
was shown previously. The Cartan forms over the map are the
pullback over the map of the Cartan forms: ωγ = γ∗ω.

The equation governing the parallel transport of w is

∇γ
∂/∂tw = 0. (0.141)

Let’s figure out what this equation is for transport of an arbitrary
vector w along an arbitrary path γ on a sphere. We start by
constructing the necessary manifolds.

(instantiate-coordinates the-real-line ’t)

(define M (rectangular 2))
(instantiate-coordinates M ’(theta phi))
(define M-basis (coordinate-system->basis M))

We specify a connection by giving the Christoffel symbols.37

37We will show how to get these Christoffel symbols from a metric.

(define G-S2-1
(make-Christoffel
(let ((zero (lambda (point) 0)))
(down (down (up zero zero)

(up zero (/ 1 (tan theta))))
(down (up zero (/ 1 (tan theta)))

(up (- (* (sin theta) (cos theta))) zero))))
M-basis))

Next, we need the path γ, which we represent as a map from the
real line to M , and w, the parallel-transported vector over the
map:

(define gamma:N->M
(compose (M ’->point)

(up (literal-function ’alpha)
(literal-function ’beta))

(the-real-line ’->coords)))

where alpha is the colatitude and beta is the longitude.

(define basis-over-gamma
(basis->basis-over-map gamma:N->M M-basis))

(define w
(basis-components->vector-field
(up (compose (literal-function ’w0)

(the-real-line ’->coords))
(compose (literal-function ’w1)

(the-real-line ’->coords)))
(basis->vector-basis basis-over-gamma)))

(define sphere-Cartan-over-gamma
(Christoffel->Cartan-over-map G-S2-1 gamma:N->M))

Finally, we compute the residual of the equation (0.140) that gov-
erns parallel transport for this situation:



58

(print-expression
(s:map/r
(lambda (omega)
((omega

(((covariant-derivative sphere-Cartan-over-gamma)
d/dt)
w))

((the-real-line ’->point) ’tau)))
(basis->1form-basis basis-over-gamma)))

(up (+ (* -1
(sin (alpha tau))
(cos (alpha tau))
((D beta) tau)
(w1 tau))

((D w0) tau))
(/ (+ (* (w0 tau) (cos (alpha tau)) ((D beta) tau))

(* ((D alpha) tau) (cos (alpha tau)) (w1 tau))
(* ((D w1) tau) (sin (alpha tau))))

(sin (alpha tau))))

Thus the equations governing the evolution of the components of
the transported vector are:

Dw0(τ) = sin(α(τ)) cos(α(τ))Dβ(τ)w1(τ)

Dw1(τ) = −cos(α(τ))
sin(α(τ))

(
Dβ(τ)w0(τ) + Dα(τ)w1(τ)

)
(0.142)

These equations describe the transport on a sphere, but more
generally they look like

Dw(τ) = f(σ(τ),Dσ(τ))w(τ), (0.143)

where σ is the tuple of the coordinates of the path on the manifold
and w is the tuple of the components of the vector. The equation is
linear in w and driven by the path σ, as in a variational equation.

Suppose that the path is not known explicitly, but rather that
it is the integral curve of a vector field v on the manifold. If b
is the tuple of components of v (recall that b = (v(χ)) ◦ χ−1, see
equation 0.26) then the evolution of σ and w is governed by

Dσ(τ) = b(σ(τ)) (0.144)
Dw(τ) = f(σ(τ),Dσ(τ))w(τ), (0.145)

which we can collectively write as

Ds(τ) = g(s(τ)), (0.146)

by defining the state s = (σ,w) and substituting the right-hand
side of equation (0.144) into equation (0.145). We can evolve
this state by exponentiating the vector field defined by g on the
Euclidean space of twice the dimension of our manifold.

Geodesic Motion

In geodesic motion the velocity vector is parallel transported by
itself.38 Recall (equation 0.98) that the velocity is the differential
of the vector ∂/∂t over the map γ. More precisely, the equation
of geodesic motion is

∇γ
∂/∂tdγ(∂/∂t) = 0. (0.147)

In coordinates, this is

D2σi(t) −
∑
jk

Γi
jk(γ(t))Dσj(t)Dσk(t) = 0, (0.148)

where σ(t) is the coordinate path corresponding to the manifold
path γ.

For example, let’s consider geodesic motion on the surface of a
unit sphere. We let

(define gamma:N->M
(compose (M ’->point)

(up (literal-function ’alpha)
(literal-function ’beta))

(the-real-line ’->coords)))

38 The geodesic equation is often written

∇vv = 0,

but this is nonsense. The geodesic equation is a constraint on the path, but
the path does not appear in this equation. Further, the velocity along a path
is not a vector field, so it cannot appear in either argument to the covariant
derivative.

be a map from the real line to the sphere, with colatitude alpha
and longitude beta. The geodesic equation is:

(print-expression
(((((covariant-derivative sphere-Cartan-over-gamma) d/dt)

((differential gamma:N->M) d/dt))
(M ’->coords))
((the-real-line ’->point) ’t)))⎛⎜⎝− cos (α (t)) sin (α (t)) (Dβ (t))2 + D2α (t)

2Dβ (t) cos (α (t))Dα (t)
sin (α (t))

+ D2β (t)

⎞⎟⎠
Notice that the program for ∇γ uses the program for ∇ but uses
a basis over the map and Cartan forms over the map for the basis
and Cartan forms.

The geodesic equation is the same as the Lagrange equation
for free motion constrained to the surface of the unit sphere. The
Lagrangian for motion on the sphere is the composition of the free
particle Lagrangian and the state transformation induced by the
coordinate constraint:39

(define (Lfree s)
(* 1/2 (square (velocity s))))

(define (sphere->R3 s)
(let ((q (coordinate s)))
(let ((theta (ref q 0)) (phi (ref q 1)))

(up (* (sin theta) (cos phi))
(* (sin theta) (sin phi))
(cos theta)))))

(define Lsphere
(compose Lfree (F->C sphere->R3)))

Then the Lagrange equations are:

39The method of formulating a system with constraints by composing a free
system with the state-space coordinate transformation that represents the
constraints can be found in SICM, Section 1.6.3.

(print-expression
(((Lagrange-equations Lsphere)
(up (literal-function ’alpha)

(literal-function ’beta)))
’t))[ − (Dβ (t))2 sin (α (t)) cos (α (t)) + D2α (t)

2Dα (t)Dβ (t) sin (α (t)) cos (α (t)) + D2β (t) (sin (α (t)))2

]

The Lagrange equations are true of the same paths as the geodesic
equations. The second Lagrange equation is the second geodesic
equation multiplied by (sin(θ0(t)))2, and the Lagrange equations
are arranged in a down tuple, whereas the geodesic equations are
arranged in an up tuple. The two systems are equivalent unless
θ0(t) = 0, where the coordinate system is singular.40

Curvature

If the intrinsic curvature is not zero a vector parallel transported
around a small loop will end up different from the vector that
started. The Riemann tensor encapsulates this idea.

The Riemann curvature operator is

R(u, v) = [∇u,∇v] −∇[u,v] (0.149)

or

(define ((Riemann-curvature Cartan) u v)
(let ((nabla (covariant-derivative Cartan)))
(- (commutator (nabla u) (nabla v))

(nabla (commutator u v)))))

This allows us to compute the intrinsic curvature of a manifold
at a point. The Riemann curvature takes a vector field as an
argument and returns a vector field as a result.

The traditional Riemann tensor is

R(ω, x, u, v) = ω((R(u, v))(x)), (0.150)

40The geodesic equations and the Lagrange equations are related by a con-
traction with the metric.

where ω is a one form field that measures the negative of the incre-
mental change in the vector field x caused by parallel transporting
it around the loop defined by the vector fields u and v.

So, for example,

(define sphere-Cartan (Christoffel->Cartan G-S2-1))

(print-expression
(((Riemann sphere-Cartan) dphi d/dtheta d/dphi d/dtheta)
((M ’->point) (up ’theta0 ’phi0))))

+1

Here we have computed the φ component of the result of carrying
a ∂/∂θ basis vector around the parallelogram defined by ∂/∂φ and
∂/∂θ. The result shows a net rotation in the φ direction.

Most of the coefficients of the Riemann tensor for the sphere
are zero. The following are the nonzero coefficients:

R

(
dθ,

∂

∂φ
,

∂

∂θ
,

∂

∂φ

)
(χ−1(qθ, qφ)) =

(
sin(qθ)

)2
R

(
dθ,

∂

∂φ
,

∂

∂φ
,

∂

∂θ

)
(χ−1(qθ, qφ)) = −

(
sin(qθ)

)2
R

(
dφ,

∂

∂θ
,

∂

∂θ
,

∂

∂φ

)
(χ−1(qθ, qφ)) = −1

R

(
dφ,

∂

∂θ
,

∂

∂φ
,

∂

∂θ

)
(χ−1(qθ, qφ)) = 1 (0.151)

Explicit Transport
The result of the Riemann calculation is what we expect by ex-
plicitly doing the transport.

Equation (0.145) governs parallel transport along an integral
curve σ of the vector field v with coefficient function b. The col-
lective equation (0.146) is

Ds(τ) = g(s(τ)), (0.152)

where s = (σ,w). The differential equations describing the evolu-
tion of a function h of state s along the state path are

D(h ◦ s) = (Dh ◦ s)(g ◦ s) = Lgh ◦ s. (0.153)

Exponentiation gives a finite evolution:

h(s(τ + ε)) = (eεLgh)(s(τ)). (0.154)

We want to parallel transport a vector w around a parallel-
ogram defined by two vectors u and v. The vector w is really a
vector over a map, where the map is the parametric curve describ-
ing our parallelogram. This map is implicitly defined in terms of
the vector fields u and v. Let gu and gv be the right-hand sides
of the differential equations for parallel transport along u and v
respectively. Then evolution along u for interval ε, then along v
for interval ε, then reversing u, and reversing v, brings σ back to
where it started to second order in ε.

The state s = (σ,w) after transporting s0 around the loop is

(e−εLgv I) ◦ (e−εLgu I) ◦ (eεLgv I) ◦ (eεLgu I)(s0)
= (eεLgu eεLgv e−εLgu e−εLgv I)(s0)

= (eε2[Lgu ,Lgv ]+···I)(s0). (0.155)

So the lowest order change in the transported vector is

ε2W (([Lgu
, Lgv

]I)(s0)), (0.156)

where W (σ,w) = w.
Recall the differential equations (0.142) for parallel transport

by the vector v of the components w of the vector w on the sphere
are

Dw0(τ) = sin(α(τ)) cos(α(τ))Dβ(τ)w1(τ)

Dw1(τ) = −cos(α(τ))
sin(α(τ))

(
Dβ(τ)w0(τ) + Dα(τ)w1(τ)

)
,

where α and β are functions giving the coordinates of the path
for each value of the parameter τ .

To integrate these equations of the coordinates of the vector be-
ing transported along a path σ(τ) = (α(τ), β(τ)), defined by dif-
ferential equations we need to make a state space that represents
both the path and the coordinates of the vector being transported.
To represent these states we make a new manifold with four coor-
dinates. The first two coordinates are the coordinates of the path.
The second two coordinates are the components of the vector to

be transported, relative to the coordinate directions in the original
manifold M .41 The states are s = (σ,w) = ((θ, φ), (w0, w1))

(define M (rectangular 2))
(instantiate-coordinates M ’(theta phi))

(define states (rectangular 4))
(instantiate-coordinates states ’(Theta Phi w0 w1))

The differential equations (0.144) for the path are Dσ(τ) =
b(σ(τ)). The differential equations for the coordinates of the vec-
tor are driven by the path σ. The right-hand side of this composite
differential equation is a vector field on this manifold. The vector
field for a path through this bigger state space, given a constant
vector field v for the path segment is constructed by:

(define (G v)
(let ((alphadot (dTheta v)) (betadot (dPhi v)))
(+ v

(* (compose (* sin cos) Theta) betadot w1 d/dw0)
(* -1

(compose (/ cos sin) Theta)
(+ (* w0 betadot) (* w1 alphadot))
d/dw1))))

We can make vector fields which drag a vector in the θ and the φ
directions by specifying the components of the vector field.

(define Gu (G d/dTheta))
(define Gv (G d/dPhi))

We also need to make initial states for the integration. These
are the initial coordinates, and the initial vector to be transported.

(define (initial-state initial-coords w)
(let ((Theta0 (ref initial-coords 0))

(Phi0 (ref initial-coords 1)))
(let ((m ((M ’->point) (up Theta0 Phi0))))

((states ’->point)
(up Theta0 Phi0 ((dtheta w) m) ((dphi w) m))))))

41In the programs below there are both capitalized and lower-case theta and
phi. This is to avoid a symbol conflict among the coordinate functions defined
for the two manifolds. We need to distinguish them.

If we project the commutator of these vector fields in the θ and
φ directions, applied to initial vectors that are in the φ and θ
directions we get the same discrepancy predicted by the Riemann
tensor!

(print-expression
((dw0 (commutator Gu Gv))
(initial-state (up ’Theta0 ’Phi0) d/dphi)))

(* -1 (expt (sin Theta0) 2))

(print-expression
((dw1 (commutator Gu Gv))
(initial-state (up ’Theta0 ’Phi0) d/dtheta)))

1

Torsion
There are many connections that describe the local properties
of any particular manifold. A connection has a property called
torsion, which is computed as follows:

T (u, v) = ∇uv −∇vu − [u, v] (0.157)

The torsion takes two vector fields and produces a vector field. As
a program this shows the dependence on the Cartan forms, which
specify the connection:

(define ((torsion Cartan) u v)
(let ((nabla (covariant-derivative Cartan)))
(+ ((nabla u) v)

(* -1 ((nabla v) u))
(* -1 (commutator u v)))))

The torsion for the connection for the 2-sphere specified by the
Christoffel symbols G-S2-1 above is zero. We demonstrate this by
applying the torsion to the basis vector fields:

(for-each
(lambda (x)

(for-each
(lambda (y)

(print-expression
((((torsion sphere-Cartan) x y) a-function)
((M ’->point) (up ’theta0 ’phi0)))))

(list d/dtheta d/dphi)))
(list d/dtheta d/dphi))

0
0
0
0

Metrics

We often want to impose further structure on a manifold to allow
us to define lengths and angles. This is done by generalizing the
idea of the Euclidean dot product, which allows us to compute
lengths of vectors and angles between vectors in traditional vector
algebra.

For vectors �u = uxx̂ + uyŷ + uzẑ and �v = vxx̂ + vyŷ + vz ẑ the
dot product is �u · �v = uxvx + uyvy + uzvz. The generalization is
to provide coefficients for these terms and to include cross terms,
consistent with the requirement that the function of two vectors is
symmetric. This symmetric, bilinear, real-valued function of two
vector fields is called a metric field. Given a metric field g and
a vector field v there is a scalar field g(v, v) that is the squared
length of the vector at each point of the manifold.

For example, the natural metric on a sphere of radius R is

g(u, v) = R2(dθ(u)dθ(v) + (sin ◦ θ)2dφ(u)dφ(v)), (0.158)

and the Lorentz metric on the four-dimensional space of special
relativity is

g(u, v) = dx(u)dx(v)+dy(u)dy(v)+dz(u)dz(v)−c2dt(u)dt(v).(0.159)

Although these examples are expressed in terms of a coordinate
basis, the metric is invariant under coordinate transformations.

The metric can be used to construct a one-form field ωu from a
vector field u, such that for any vector field v we have

ωu(v) = g(v, u). (0.160)

The operation of constructing a one-form field from a vector field
using a metric is called “lowering” the vector field.

There is also an inverse metric that takes two one-form fields.
It is defined by the relation

δi
k =

∑
j

g−1(ẽi, ẽj)g(ej , ek), (0.161)

where e and ẽ are any basis and its dual basis.
The inverse metric can be used to construct a vector field vω

from a one-form field ω, such that for any one-form field τ we have

τ(vω) = g−1(ω, τ). (0.162)

The operation of constructing a vector field from a one-form field
using a metric is called “raising” the one-form field ω.

This vector field can be explicitly computed from the one-form
field, with respect to a basis as follows:

vω =
∑

i

g−1(ω, ẽi)ei (0.163)

The raising and lowering operations allow one to interchange
the vector fields and the one-form fields. However they should not
be confused with the dual operation that allows one to construct a
dual one-form basis from a vector basis or construct a vector basis
from a one-form basis. The dual operation that interchanges bases
is defined without assigning a metric structure on the space.

There is a unique torsion-free connection induced by a metric.
The Christoffel coefficients of the first kind are computed from the
metric by the following

Γ̄ijk =
1
2
(ek(g(ei, ej)) + ej(g(ei, ek)) − ei(g(ej , ek))), (0.164)

for the coordinate basis e. We can then construct the Christoffel
coefficients of the second kind (the ones used previously to define

a connection) by “raising the first index.” To do this we define a
function of three vectors, with a wierd currying:

Γ̃(v,w)(u) =
∑
ijk

Γ̄ijkẽ
i(u)ẽj(v)ẽk(w) (0.165)

This function takes two vector fields and produces a one-form field.
We can use it with equation (0.163) to construct a new function
that takes two vector fields and produces a vector field:

Γ̂(v,w) =
∑

i

g−1(Γ̃(v,w), ẽi)ei (0.166)

We can now construct the Christoffel coefficients of the second
kind:

Γi
jk = ẽi(Γ̂(ej , ek)) =

∑
m

Γ̄mjkg
−1(ẽm, ẽi) (0.167)

The Cartan forms are then just

ωi
j =

∑
k

Γi
jkẽ

k =
∑

k

ẽi(Γ̂(ej , ek))ẽk (0.168)

So, for example, we can compute the Christoffel coefficients for
the sphere from the metric for the sphere. First, we need the
metric

(define 2-sphere (rectangular 2))
(instantiate-coordinates 2-sphere ’(theta phi))
(define 2-sphere-basis (coordinate-system->basis 2-sphere))

(define ((g-sphere R) u v)
(* (square R)

(+ (* (dtheta u) (dtheta v))
(* (compose (square sin) theta)

(dphi u)
(dphi v)))))

The Christoffel coefficients of the first kind are a complex structure
with all three indices down

(print-expression
((Christoffel->symbols

(metric->Christoffel-1 (g-sphere ’R) 2-sphere-basis))
((2-sphere ’->point) (up ’theta0 ’phi0))))

(down
(down (down 0 0)

(down 0 (* (* (cos theta0) (sin theta0)) (expt R 2))))
(down (down 0 (* (* (cos theta0) (sin theta0)) (expt R 2)))

(down (* (* -1 (cos theta0) (sin theta0)) (expt R 2))
0)))

And the Christoffel coefficients of the second kind have the inner-
most index up

(print-expression
((Christoffel->symbols

(metric->Christoffel-2 (g-sphere ’R)
2-sphere-basis)))

((2-sphere ’->point) (up ’theta0 ’phi0))))
(down (down (up 0 0)

(up 0 (/ (cos theta0) (sin theta0))))
(down (up 0 (/ (cos theta0) (sin theta0)))

(up (* -1 (cos theta0) (sin theta0)) 0)))

Hodge Star

The vector space of p-form fields on an n-dimensional manifold
has dimension n!/((n − p)!p!). This is the same dimension as the
space of n–p-form fields. So these vector spaces are isomorphic. If
we have a metric there is a natural isomorphism; for each p-form
field ω on an n-dimensional manifold with a metric there is a n–
p-form field ∗ω, called its Hodge dual. The Hodge dual should not
be confused with the duality of vector bases and one-form bases,
which is defined without reference to a metric. The Hodge dual is
useful for the elegant formalization of electrodynamics.

In Euclidean 3-space, if we think of a one form as a foliation of
the space, then the dual is a two form of two-dimensional tubes,
whose axes are perpendicular to the leaves of the foliation. The
original one-form divides these tubes up into volume elements.
For example, the dual of the basis one form dx is the two form
∗dx = dy∧dz. We may think of dx as a set of planes perpendicular
to the x axis. Then ∗dx is a set of tubes parallel to the x axis. In
higher dimensional spaces the visualization is more complicated,

but the basic idea is the same. The Hodge dual of a two-form in
two dimensions is a two-form that is perpendicular to the given
two form. However, if the metric is indefinite (e.g. the Lorentz
metric) there is an added complication with the signs.

The Hodge dual is a linear operator, so it can be defined by
its action on the basis elements. Let { ∂

∂x0 , . . . , ∂
∂xn−1 } be an or-

thonormal basis of vector fields42 and let {dx0, . . . , dxn−1} be the
ordinary dual basis for the one-forms, then the n–p-form ∗ω that
is the Hodge dual of the p-form ω can be defined by its coefficients
with respect to the basis, using indices, as

(∗ω)jp...jn−1

=
∑

i0...ip−1j0...jp−1

1
p!

ωi0...ip−1g
i0j0 . . . gip−1jp−1εj0...jn−1, (0.169)

where gij are the coefficients of the inverse metric and εj0...jn−1 is
either −1 or +1 if the permutation {0 . . . n− 1} → {j0 . . . jn−1} is
odd or even, respectively.

Electrodynamics

Using Hodge duals we can represent electrodynamics in an elegant
way. Maxwell’s electrodynamics is invariant under Lorentz trans-
formations. We use four-dimensional rectangular coordinates for
the flat spacetime of special relativity.

(define SR (rectangular 4))
(instantiate-coordinates SR ’(t x y z))
(define an-event ((SR ’->point) (up ’t0 ’x0 ’y0 ’z0)))
(define c ’c) ; We like units.

The Lorentz metric is

g(u, v) = dx(u) dx(v)+dy(u) dy(v)+dz(u) dz(v)−c2dt(u) dt(v)(0.170)

As a program

42We have a metric, so we can define “orthonormal” and use it to construct
an orthonormal basis given any basis. The Gram-Schmidt procedure does the
job.

(define (g-Lorentz u v)
(+ (* (dx u) (dx v))

(* (dy u) (dy v))
(* (dz u) (dz v))
(* -1 (square c) (dt u) (dt v))))

The invariant length of a vector is described in terms of the
metric.

‖v‖ = g(v, v) (0.171)

If the length is positive the vector is spacelike. If the length is
negative the vector is timelike. If the length is zero the vector is
lightlike or null.

The vector-field basis that is convenient for describing electro-
dynamics is orthonormal with respect to the metric.

(define SR-vector-basis
(down (* (/ 1 c) d/dt) d/dx d/dy d/dz))

(define SR-1form-basis
(up (* c dt) dx dy dz))

(define SR-basis
(make-basis SR-vector-basis

SR-1form-basis))

Indeed, the vector-field basis is orthonormal:

(print-expression
((SR-1form-basis SR-vector-basis)
an-event))

(up (down 1 0 0 0)
(down 0 1 0 0)
(down 0 0 1 0)
(down 0 0 0 1))

In this formulation of electrodynamics the electric and magnetic
fields are represented together as a two-form field, the Faraday ten-
sor. We will see that under Lorentz transformation the individual
components are mixed. The Faraday tensor is:43

43This representation is from Misner, Thorne, and Wheeler, Gravitation, p.108

(define (Faraday Ex Ey Ez Bx By Bz)
(+ (* Ex c (wedge dx dt))

(* Ey c (wedge dy dt))
(* Ez c (wedge dz dt))
(* Bx (wedge dy dz))
(* By (wedge dz dx))
(* Bz (wedge dx dy))))

The Hodge dual of the Faraday tensor exchanges the electric and
magnetic fields, negating the components that will involve time.
The result is called the Maxwell tensor:

(define (Maxwell Ex Ey Ez Bx By Bz)
(+ (* -1 Bx c (wedge dx dt))

(* -1 By c (wedge dy dt))
(* -1 Bz c (wedge dz dt))
(* Ex (wedge dy dz))
(* Ey (wedge dz dx))
(* Ez (wedge dx dy))))

We make a Hodge dual operator for this situation

(define SR-star
(Hodge-star g-Lorentz SR-basis))

and indeed, it transforms the Faraday tensor into the Maxwell
tensor:

(print-expression
(((- (SR-star (Faraday ’Ex ’Ey ’Ez ’Bx ’By ’Bz))

(Maxwell ’Ex ’Ey ’Ez ’Bx ’By ’Bz))
(literal-vector-field ’u SR)
(literal-vector-field ’v SR))
an-event))

0

One way to get electric fields is to have charges; magnetic fields
can arise from motion of charges. In this formulation we combine
the charge density and the current to make a 4-vector field:

(define (J charge-density Ix Iy Iz)
(- (* (/ 1 c) (+ (* Ix dx) (* Iy dy) (* Iz dz)))

(* charge-density c dt)))

The placement of c makes the components of the 4-vector uniform
with respect to units.

To develop Maxwell’s equations we need a general Faraday field
and a general current-density field:

(define F
(Faraday (literal-manifold-function ’Ex SR)

(literal-manifold-function ’Ey SR)
(literal-manifold-function ’Ez SR)
(literal-manifold-function ’Bx SR)
(literal-manifold-function ’By SR)
(literal-manifold-function ’Bz SR)))

(define 4-current
(J (literal-manifold-function ’rho SR)

(literal-manifold-function ’Ix SR)
(literal-manifold-function ’Iy SR)
(literal-manifold-function ’Iz SR)))

Maxwell’s equations in this language are

dF = 0 (0.172)
d(∗F) = 4π ∗ J. (0.173)

The first equation gives us what would be written in vector nota-
tion as

div �B = 0, (0.174)

and

curl �E = −1
c

d �B

dt
. (0.175)

The second equation gives us what would be written in vector
notation as

div �E = 4πρ, (0.176)

and

curl �B =
1
c

d �E

dt
+

4π
c

�I. (0.177)

To see how these work out, we must evaluate each component.
Since Faraday is a two-form field its exterior derivative is a three-
form field, so we have to provide three basis vectors; so there
are four possibilities for each equation; each possibility will yield

one of Maxwell’s equations, written in coordinates, without vector
notation.

(print-expression
(((d F) d/dx d/dy d/dz) an-event))

(+ (((partial 1) Bx) (up t0 x0 y0 z0))
(((partial 2) By) (up t0 x0 y0 z0))
(((partial 3) Bz) (up t0 x0 y0 z0)))

∂Bx

∂x
+

∂By

∂y
+

∂Bz

∂z
= 0 (0.178)

(print-expression
(((d F) (* (/ 1 c) d/dt) d/dy d/dz) an-event))

(+ (((partial 2) Ez) (up t0 x0 y0 z0))
(* -1 (((partial 3) Ey) (up t0 x0 y0 z0)))
(/ (((partial 0) Bx) (up t0 x0 y0 z0)) c))

∂Ez

∂y
− ∂Ey

∂z
=

1
c

∂Bx

∂t
(0.179)

(print-expression
(((d F) (* (/ 1 c) d/dt) d/dz d/dx) an-event))

(+ (((partial 3) Ex) (up t0 x0 y0 z0))
(* -1 (((partial 1) Ez) (up t0 x0 y0 z0)))
(/ (((partial 0) By) (up t0 x0 y0 z0)) c))

∂Ex

∂z
− ∂Ez

∂x
=

1
c

∂By

∂t
(0.180)

(print-expression
(((d F) (* (/ 1 c) d/dt) d/dx d/dy) an-event))

(+ (((partial 1) Ey) (up t0 x0 y0 z0))
(* -1 (((partial 2) Ex) (up t0 x0 y0 z0)))
(/ (((partial 0) Bz) (up t0 x0 y0 z0)) c))

∂Ey

∂x
− ∂Ex

∂y
=

1
c

∂Bz

∂t
(0.181)

(print-expression
(((- (d (SR-star F)) (* ’4pi (SR-star 4-current)))

d/dx d/dy d/dz)
an-event))

(+ (* -1 4pi (rho (up t0 x0 y0 z0)))
(((partial 1) Ex) (up t0 x0 y0 z0))
(((partial 2) Ey) (up t0 x0 y0 z0))
(((partial 3) Ez) (up t0 x0 y0 z0)))

∂Ex

∂x
+

∂Ey

∂y
+

∂Ez

∂z
= 4πρ (0.182)

(print-expression
(((- (d (SR-star F)) (* ’4pi (SR-star 4-current)))

(* (/ 1 c) d/dt) d/dy d/dz)
an-event))

(+ (/ (* 4pi (Ix (up t0 x0 y0 z0))) c)
(* -1 (((partial 2) Bz) (up t0 x0 y0 z0)))
(((partial 3) By) (up t0 x0 y0 z0))
(/ (((partial 0) Ex) (up t0 x0 y0 z0)) c))

∂By

∂z
− ∂Bz

∂y
= −1

c

∂Ex

∂t
− 4π

c
Ix (0.183)

(print-expression
(((- (d (SR-star F)) (* ’4pi (SR-star 4-current)))

(* (/ 1 c) d/dt) d/dz d/dx)
an-event))

(+ (/ (* 4pi (Iy (up t0 x0 y0 z0))) c)
(* -1 (((partial 3) Bx) (up t0 x0 y0 z0)))
(((partial 1) Bz) (up t0 x0 y0 z0))
(/ (((partial 0) Ey) (up t0 x0 y0 z0)) c))

∂Bz

∂x
− ∂Bx

∂z
= −1

c

∂Ey

∂t
− 4π

c
Iy (0.184)

(print-expression
(((- (d (SR-star F)) (* ’4pi (SR-star 4-current)))

(* (/ 1 c) d/dt) d/dx d/dy)
an-event))

(+ (/ (* 4pi (Iz (up t0 x0 y0 z0))) c)
(* -1 (((partial 1) By) (up t0 x0 y0 z0)))
(((partial 2) Bx) (up t0 x0 y0 z0))
(/ (((partial 0) Ez) (up t0 x0 y0 z0)) c))

∂Bx

∂y
− ∂By

∂x
= −1

c

∂Ez

∂t
− 4π

c
Iz (0.185)

